





MSXソフト集Part.3
好評発売中



定価850円
ISBN4-416-18449-2 C2055 ¥850E

誠文堂新光社

MSXソフト集 Part.1



MSXソフト集 Part.1



MSXソフト集Part.2
好評発売中

誠文堂新光社

MSX


ソフト集

Part.1

MSX ソフト集 Part.1

誠文堂新光社

MSXソフト集 Part.1



MSX


ソフト集 Part I

■プログラムの打ち込み方, ロード,セーブのし方	4
■MSX-BASICの特長	10
■ブロックくずし	18
■ホームデータベース	27
■スーパーバックアップ	35

イラスト：窪野計子 表紙デザイン：JUN KIKKAWA 表紙イラストレーション：ECOV

もくじ

■コンピュレコーダーII	42
■スーパーメイズ	52
■ハイパーチェイサー	59
■カセットファイルズ	69
■ balan	77
■15ゲーム	91
■もぐらたたき	99



BASICの打ち込み方 と ロード・セーブの仕方

これから紹介するソフトは、すべて BASIC で作っていますが、この打ち込み方とロード、セーブについて話したいと思います。

なんといってもパソコンは、プログラムがなければただのハコですね。そこでプログラムの打ち込みですが、まずリストを見てください。左はじに数字が10ごとに並んでいます、この数字を行番号といいます。プログラムの入力、この行番号ごとに横に並んだ文字を入れていくのです。

たとえばリスト1のようなプログラムの場合、

`1`、`0`、`スペース`、`SHIFT` + `7`、…… `RETURN`

で1行目が終了です。続いて、20、30…と入力していきますが、ひとつの行番号に始まるプログラムを打ち込んだら、かならず最後にリターンキーを押して、パソコンに登録してください。これをわすれると、たとえば20で始まる行番号のプログラムがどこで終るか、パソコンがわからなくなるわけです。

AUTO

ところで、いちいち行番号を入力するのはめんどうだと思いませんか。そんなときに便利な命令に **AUTO** というのがあります。MSXの電源を入れると、画面下に5種類の命令が表示されてますね。これはキーボードにあるF1、F2…というファンクションキーといわれるキーに対応していますが、このF2のキーに **AUTO** 命令が登録されています。このF2を押してリターンキーを押せば実行されます。もちろん、自分で`␣`、`␣`、`␣`、`␣`と打ち込んでもOKで、この命令が実行されると、画面に10という数字が表示されますので、先ほど話した方法で、このあとプログラムを打ち込めば良いわけです。

リストI

```
10 ' Block くずし
20 ' by Tamae Tama
30 ' debug and advice by Haruka takagi
40 ' copyright 1984 (C) NATS
50 ' copyright 1984 (C)
60 ' Seibundo-Shinkousha publishing
70 '
80 ' HENSU & SYSTEM INIT
90 '
100 MAXFILES=3: CLEAR 100, &HCFFF: DEFINT A-
    000
110 DEFUSR1=&HD010: DEFUSR2=&HD021: OPEN "GR
    T AS #1
120 XS=0: YS=0: 'BALL SPEED X,Y
130 XB=0: YB=0: 'BALL X,Y
140 XP=0: ' PADL X
150 TP=0: ' HI- SCORE
160 SCREEN 1:CLS
```

それから、この **AUTO** 命令が働いているときにリストを全部打ち終ったらどうするかというと、いくらリターンキーを押しても、どんどん数字が出てくるだけです。このときは、最後の行を打ち込んでリターンキーを押したら、次に **CTRL** キーと **STOP** キーをいっしょに押せば、**AUTO** 命令が解除になります。これをブレークといいます。

ところで、リストの行番号がずーっと順序よく並んでいれば良いのですが、途中で飛んでいることもありますね。たとえば510行の次が1000だったりなんてときに **AUTO** で打ち込んでいるときはどうするんじゃ、なんてことになりますね。このときは、510を打ち終り、リターンキーを押したら、ブレーク (**CTRL**+**STOP**) で一度 **AUTO** 命令を終らせて、次に **AUTO** 1000と打ち込めば1000番地から同じように続けて打ち込めますよ。もちろんリストを全部打ち込みが終ったら、先ほどと同じようにブレーク (**CTRL**+**STOP**) で終了とします。

ところで、本書のリストは、ちょっと皆さんが見なれたものとは違うと思います。これは、リストの行番号とプログラムを見やすく分けるためにこのようにしています。もしあなたが打ち込むときに

1つの行番号が2行や3行になるときは、行番号の下のところまで右はしにいっぱいプログラムを打ち込んでください。実際には、リスト 3, 4 のようになりますよ。

これでプログラムが打ち終わったら、こんどはバグ取りです。バグ取りってなんだろうって思っている人も多いと思いますが、ようするに、プログラムの打ち込みの間違いがないように確認することなのです。もちろん、実際に走らせてやる方法もありますが、まず打ち込んだリストをよく見なければなりません。このとき使う命令に **LIST** があります。

LIST

LIST 命令は、打ち込んだリストを画面に表示させる命令です。この命令も、ファンクションキーの F 4 に入っていますので、こちら

リスト 2

```
820 'display all data
830 IF P=0 THEN PRINT :PRINT :PRINT "また テータ か" ありませ
    ン !! ":BEEP:PRINT :GOTO 850
840 PRINT :FOR L1=1 TO P:FOR L2=0 TO 5:PRINT USING "テ
    ータ (#)";L2+1;:PRINT D$(L2,L1):NEXT :FOR W=0 TO 5
    00:NEXT W:PRINT :NEXT
850 PRINT "つき"のしよりを するために なにか キー を おして くだ"さい !"
860 IF INKEY$="" THEN 860 ELSE 210
870 'end of job
880 PRINT :PRINT :PRINT "こ"うさまで"した":ON ERROR GOTO 0
    :END
890 PRINT :FOR L=0 TO 5:PRINT USING "テータ (#) キーワート"
    ;L+1;:KW$(L)="" :INPUT KW$(L):NEXT:RETURN
900 KA=1:FOR L1=0 TO 5:KK=INSTR(D$(L1,KS),KW$(L1)):KA
    =KA*KK:NEXT
910 IF KA=0 THEN KL=0:RETURN
920 PRINT :PRINT "***** この テータ で"すか ? *****":PRINT :
    FOR L1=0 TO 5:PRINT USING "テータ (#) " ;L1+1;:PRI
    NT D$(L1,KS):NEXT :KL=1:RETURN
```

リスト 3

```
810 PRINT :INPUT "また つづ"けますか ";A$:IF A$="y" OR A$="Y" THEN 780 ELSE 210
820 'display all data
830 IF P=0 THEN PRINT :PRINT :PRINT "また テータ か" ありません !! ":BEEP:PRINT :GOTO 850
840 PRINT :FOR L1=1 TO P:FOR L2=0 TO 5:PRINT USING "テータ (#)";L2+1;:PRINT D$(L2,
    L1):NEXT :FOR W=0 TO 500:NEXT W:PRINT :NEXT
850 PRINT "つき"のしよりを するために なにか キー を おして くだ"さい !"
860 IF INKEY$="" THEN 860 ELSE 210
870 'end of job
880 PRINT :PRINT :PRINT "こ"うさまで"した":ON ERROR GOTO 0:END
890 PRINT :FOR L=0 TO 5:PRINT USING "テータ (#) キーワート" ;L+1;:KW$(L)="" :INPUT KW$(L
    ):NEXT:RETURN
900 KA=1:FOR L1=0 TO 5:KK=INSTR(D$(L1,KS),KW$(L1)):KA=KA*KK:NEXT
910 IF KA=0 THEN KL=0:RETURN
920 PRINT :PRINT "***** この テータ で"すか ? *****":PRINT :FOR L1=0 TO 5:PRINT USING "
    テータ (#) " ;L1+1;:PRINT D$(L1,KS):NEXT :KL=1:RETURN
930 CLS
940 PRINT "エラー か" はっせい しました!!!"
950 PRINT "もういちど" やりなおしてくだ"さい"
960 INPUT "なにか キーを おしてくだ"さい"
```


リスト 4
プリントしてありますが、モニター画面にはこのように出る。

```
820 'display all data
830 IF P=0 THEN PRINT :PRINT
:PRINT "また データ か" ありません !!
":BEEP:PRINT :GOTO 850
840 PRINT :FOR L1=1 TO P:FOR
L2=0 TO 5:PRINT USING "データ (
#)" ;L2+1 ;:PRINT D$(L2,L1):NEX
T :FOR W=0 TO 500:NEXT W:PRIN
T :NEXT
850 PRINT "つき" の しりを するために な
か キー を おして くだい !"
860 IF INKEY$="" THEN 860 ELS
E 210
870 'end of job
```

を押す方が早くて良いでしょう。

この命令を実行すると、打ち込んだリストが、次から次へと画面に表示され、とても確認するなんてむりですが、このときは **STOP** キーを押してください。その場で画面が一時停止になりますので、このときにリストを確認します。

もしリストに打ち込み間違いがあったら、ブレイクして(CTRL+STOP キーで止める) からカーソルキーで間違ったところにカーソルを移動させて、正しいリストを打ち込みます。もしブレイクさせないでカーソルを動かそうとしても動きませんし、画面にカーソルも表示されてませんよ。

間違いがなければ、再び STOP キーを押せば、リストの続きが表示されますので、同じ方法で確認してください。もしプリンターやプロッターを持っていれば、**LLIST** 命令を入力すれば、リストが印字されますので、それで確認してください。

確認が終わったらすぐに RUN させたいところですが、まだまだあせってはいけません。打ち込んだリストを、まずカセットなりディスクに記録しておきましょう。これは、BASIC で書かれたプログラムでも、マシン語を DATA 文として使って、実際に動くときはマシン語で動くプログラムもありますので、SAVE しておきます。このことを **バックアップ** を取るといいます。

マシン語で動いているパソコンは、一度 RUN させてしまうと先

**SAVE
CSAVE
BSAVE**

ほどまで使えたブレーク (CTRL+STOP キー) が使えなくなります。ようするに、ROM カートリッジのソフトを使っているのと同じなのです。こうなると、リセットボタンを押すか、電源スイッチを切らないと止まらなくなります。ということは、間違いなく打ち込んだプログラムもその場でパーと消えちゃうのです。

また、ちゃんとプログラムが走ればまだ良いのですが、間違いがあるとまったくウンもスンもいってくれず、何もできないこともあります。このときもリセットや電源のお世話にならなければならないので、プログラムはパーですね。ですから、かならず打ち込みと確認が終わったらセーブしておく必要があるのです。

というわけで、続いて記録する命令に話をすすめます。

このプログラムを記録しておく命令には、**SAVE**、**CSAVE**、**BSAVE** と3種類ありますが、**BSAVE** という命令はマシン語のプログラムの場合に使用しますので、今回の BASIC のプログラムでは必要ありません。

この **SAVE**、**CSAVE** は、どちらを使っても良いのですが、読み出す方法も、記録方法により決まります。

まずカセットテープに記録する場合のポピュラーな方法は、**CSAVE** 命令です。使い方は、

CSAVE "broker"

という形で、最後にリターンキーを押せば、記録が始まります。この“(ダブルコーテーション) で囲まれた6文字はファイルネームと

<pre>CSAVE "b l o c k " OK CLOAD? FOUND: b l o c k OK CLOAD "b l o c k " FOUND: b l o c k OK</pre>	<pre>SAVE "CAS: b l o c k " OK LOAD "CAS: b l o c k " FOUND: b l o c k OK LOAD "CAS: b l o c k ", r FOUND: b l o c k</pre>
--	--

いて、プログラムの名前が入ります。このファイルネームは6文字以内でつけます。

CLOAD?

記録が終わったら、テープを録音のスタート地点まで巻き戻し、再生にし、**CLOAD?** 命令を打ち込んでリターンキーを押してください。これは、プログラムがちゃんとテープに記録されたかを確認するための命令で、ベリファイとも呼ばれるものです。実行すると、

```
Found "(ファイルネーム)"
```

と表示され、最後にOKが出れば完全に記録されています。

CLOAD

読み出し方法は、もちろん**CLOAD**です。

ところでもうひとつの記録方法の**SAVE** 命令ですが、これはディスクを使うときにも同じ形で使いますが、この命令をカセットテープに使う場合は、次のような形になります。

```
SAVE "CAS : broker"
```

もしCAS : を入れないと、ディスクに記録するときの命令になってしまいます。もちろんリターンキーも押さないとだめです。このときの読み出しは、

```
LOAD "CAS : broker"
```


となり、最後の"でくくられた後に、Rとすると、読み込みが終わったらすぐスタートするという命令になります。

この**CSAVE**と**SAVE "CAS :** は、同じカセットテープに記録するのにどんな違いがあるのかと思いませんか。これは、同じカセットに記録するのでも、記録される信号の形が異なるのです。**CSAVE**は、パソコンがすぐわかる形のバイナリーと呼ばれる形式で、**SAVE**は、打ち込んだリストと同じ形で記録されるアスキーと呼ばれる形なのです。2つをくらべると、同じプログラムを記録するにしても**CSAVE**の方が短くできるのですが、**SAVE**を使うと、後からプログラム同士のドッキングができる(**MARGE** 命令を使う)のです。

MARGE

このような違いがあるわけですが、それぞれメリットを持っていますが、皆さんが実際に使われるときは、**CSAVE**の方が便利でしょう。

以上がセーブ、ロードのやり方です。これさえ覚えていれば、本書のゲームは君のもの。さあ、打ち込んで楽しんでください。



MSX

BASICの特長

MSX-BASICでゲームなどのプログラムを作るときにぜひ知っておかなければならない、いくつかの特長があるので解説しておきましょう。

MSX-BASICの3大特長は、

- ①グラフィック（特にスプライト機能とDRAW命令）
- ②PSGによるサウンド機能
- ③豊富な割り込み機能

の3つと言えます。この3つ以外は、NECのPCシリーズや、富士通のFMシリーズなどに使用されているマイクロソフトBASICとほとんど同じですので、この3つの機能さえきちんと理解していれば、他の機種用に開発されたBASICプログラムの移植も楽になると思います。

次に、この3つの命令について詳しく解説しましょう。

1. グラフィック機能


MSXの強力なグラフィック機能は、テキサス社の強力なLSI、**TMS9918A**によって可能になっています。このLSIは、別名VDP（ビデオ・ディスプレイ・プロセッサ）と呼ばれ、テレビゲームなどによく使用されているものです。

VDPの持つ強力な機能としてスプライト機能がありますが、まずこれについてお話ししましょう。

① スプライト

スプライトは、図1のように32面をかさね合わせることが可能なパターンで、ちょうどアニメーションのセル画に相当するものです。このスプライトは、アニメーションでセル画を変えて行くことにより動きを作るのと同じく、あるスプライト面に表示されるスプライトパターンを動かすことによって、動きのあるグラフィックをい

図1



も簡単に実現することが可能です。それゆえ、スプライトは動画とも呼ばれているわけです。

それでは、スプライトに関する MSX-BASIC の命令にはどんなものがあるのでしょうか？ これを表1にまとめてみました。多くないのですぐ覚えられますね。

SCREEN
SPRITE\$
PUT SPRITE

実際のプログラムでは、まず **SCREEN** 命令によってスプライトの大きさを決め、次に **SPRITE\$** 変数にスプライト・パターンの定義を行い、**PUT SPRITE** 命令で表示を行えば良いわけです。注意点としては、**SCREEN** 命令を実行すると、それまで定義していたスプライト・パターンの内容が消えてしまう点と、スプライトはテキストモードでは使用できない点、それに表示する座標をうまく画面に入るようにしないとスプライトが見えなくなってしまう点です。

さらに、同一水平方向に5つ以上のスプライトがあると、4つ目以上は表示されないので注意してください。これについては、**TMS9918A**自体がそのような仕様になっていますので、どうしようもありません。

スプライトに関する命令

SCREEN 一, 二, 三, 四, 五

二で示されるパラメーターがスプライトの大きさを示す

0: 8×8ドット標準サイズ
 1: 8×8ドット拡大サイズ
 2: 16×16ドット標準サイズ
 3: 16×16ドット拡大サイズ

但し、スプライトは、テキスト・モードでは使用できません

PUT SPRITE 一、二、三、四

一はスプライト面の番号を示します。

この値は第一図の#0～#31に対応しており、0から31までの値をとります。

二はSTEP (X, Y) または (X, Y) の表現をとるもので、スプライトの表示される場所を示しています。

STEPが付けられたときは、その座標は、前の座標との相対座標となります。

三はカラーを示しており、0から15までの値をとります。

四はスプライト番号を示しており、どのスプライト・パターンを使用するかを決めます
 このスプライト・パターンはあらかじめ、SPRITE \$変数に定義されていなければなりません

SPRITE \$

SPRITE \$変数は、スプライト・パターンの定義を行うための特殊な変数で、

SPRITE \$(1) = chr\$(1) + chr\$(2) + ...
 のように使います。

カッコの中の数字がスプライト・パターンの番号を示しており、8×8モードのときは1から255まで、16×16モードのときは、1から63までとなっています。

スプライト・パターンのデータは、以下のように作ります

●●●○○●●●	11100111	左の●と○による図形を1と0に直します これを2進数として見てデータを作ります すると、このパターンを定義するには、 以下のようにすればよいわけです
○○●○○○○●	00100001	
●●●●●●●●	11111111	
●○○●●○○●	10011001	
●●●●●●●●	11111111	
○○●○○○○●	00100100	
○○●○○○○●	00100100	
●●●○○●●●	11100111	

SPRITE \$(1) = &B11100111 + &B00100001 + &B1111
 1111 + &B10011001 + &B11111111 + &B00100100 + &B0
 0100100 + &B11100111 となるわけです

②DRAW命令

DRAW

スプライト機能と並んで重要なグラフィック命令が、**DRAW** 命令です。この命令は一口でいえば、動点（動く点）の軌跡によっていろいろな図形を描く命令で、その描きかたからタートルグラフィックなどとも呼ばれます。表 2 に **DRAW** 命令で使用されるパラメータ（サブコマンドと呼ばれます）とその使い方を示します。上手に使うと、なかなか便利なものになります。


表 2

DRAW命令のサブ・コマンド

Mx, y : 絶対位置x, yへ動点を移動する
M±x, ±y : 現在位置からx, yだけ動点を移動する
Un : 上へnだけ動点を移動する
Dn : 下へnだけ動点を移動する
Rn : 右へnだけ動点を移動する
Ln : 左へnだけ動点を移動する
En : 右上へnだけ動点を移動する
Fn : 右下へnだけ動点を移動する
Gn : 左下へnだけ動点を移動する
Hn : 左上へnだけ動点を移動する
An : 座標系をnだけ回転する
Cn : 色を指定する
Sn : 1単位のドット数を指定する

DRAW命令の使い方

例えば、今、下図のような四角形を描くことにします




まず、初点（原点）を左上とすると、右、下、左、上の順番で各4単位の線を引いて行けばよいわけです

つまり、
DRAW" R4D4L4U4"
とすれば良いわけです

この命令がどのように実行されるのかみてみましょう。
（下線の部分が現在実行している部分です）

- | | |
|------------|-----------------|
| 1. 右への線を引く | DRAW" R4D4L4U4" |
| 2. 下への線を引く | DRAW" R4D4L4U4" |
| 3. 左への線を引く | DRAW" R4D4L4U4" |
| 4. 上への線を引く | DRAW" R4D4L4U4" |



2. サウンド機能

PLAY SOUND

MSX にはサウンド用の LSI として、ゼネラルインスツルメント社の **AY-3-8910** という LSI が使用されています。この LSI は別名 PSG (プログラマブル・サウンド・ジェネレーター) と呼ばれているもので、マイコンの制御により多種多様な音を作り出すことが可能です。

MSX-BASIC では、この PSG をサポートするのに **PLAY** 命令と **SOUND** 命令の 2 つを持っていますが、初めての方は、**PLAY** 命令を使いこなすことができれば十分です。表 3 に、**PLAY** 命令のパラメーター (サブコマンド) とその使用方法を示します。この LSI は、3 重和音+ノイズ音を使用できますので、使いこなすことができれば、ゲーム作りなどがいっそう楽しくなることでしょう。

表 3

PLAY 命令のサブ・コマンド

A, B, C, D, E, F, G : 音符を示す。
音符の後に # または + が付くと半音高く、逆に - が付くと半音低くなる
たとえば、A#, B+, C- など

On : オクターブを示す (n は 1 から 8 まで)
Nn : 音程を示す (n は 0 から 96 まで)
Ln : 長さ (音楽) を示す (n は 1 から 64 まで)
Rn : 休符を示す (n は 1 から 64 まで)
Tn : テンポを示す (n は 32 から 255 まで)
Vn : 音量を示す (n は 0 から 15 まで)
Mn : エンベロープの周波数を示す (n は 1 から 65535 まで)
Sn : エンベロープのパターンを示す (n は 1 から 15 まで)
・ : 符点を示す

PLAY 命令の使い方

たとえば、ド、ミ、ソの和音を出したいときは
PLAY "C", "E", "G"
とする

和音でないときには、その音数だけのサブ・コマンドを送ればよい
PLAY "CDCDCEGFDGE", "L5R2V15CDEFG"
PLAY "AB#C-DEFGDEFG"
などである

3. 割り込み

割り込み、という言葉を知らない人もいますので、簡単に説明しておきましょう。

たとえば、あるループ状にぐるぐる回るプログラムがあったとします。このプログラムが1周するのに10秒かかるものとしますと、このプログラム内のある処理、たとえばキー入力の処理は10秒に1回しか行われないことになります。もし仮にこのゲームがリアルタイムのアクションゲームだとすると、キーの読み込みに10秒もかかっていては、とても遊ぶことはできませんね。こんなときは、今プログラムがどの行を実行していても、キーが押されたらキーの読み込み処理に実行を移し、その処理が終了したらもとの行に戻るようなことができれば、たいへんありがたいものです。

実は割り込みというのは、このように、今プログラム中のどこを実行していても、ある要因（この場合はキーを押すこと）によってすぐさま、指定された処理に移行するような処理方式を言うのです。

MSX-BASICでは、この割り込み処理が他のBASICと比較して非常に強力なものとなっており、キーによる割り込みの他にスプライトの衝突による割り込み、時間による割り込みなどがサポートされています。表4にMSX-BASICの割り込み機能をまとめておきますので参考にしてください。

表 4

MSX-BASICの割り込み機能

MSX-BASICには強力な割り込み機能がサポートされていますが、その書式はどれも同じになっています。

以下に、キーワードをxxxxとして、割り込みに関する命令を一括して説明します

ON xxxx GOSUB 行番号1、行番号2、行番号3、...

割り込みによる飛び先を指定します

飛び先の行番号がいくつかあるのは、たとえば、ファンクション・キーによる割り込みの場合など、ファンクション・キーが幾つもありますので、どのキーが押された場合どの飛び先に飛ぶのかを決めるには、キーの数だけの飛び先が必要となるからです

xxxx ON

割り込みを有効にします。つまり、指定された割り込みの処理を可能にします

xxxx OFF

割り込みを無効にします。つまり、割り込みを発生するような動作が行われても、割り込み処理は行いません

xxxx STOP

割り込みを一時、保留にします。再開は、×××× ONによって行われます

以下に、各キーワードの説明と使用方法を示します

1. ファンクション・キーによる割り込み
キーワード=KEY

F1、F2などのファンクション・キーを押すことによって発生する割り込みです
ON KEY GOSUBで指定する行番号は、それぞれ、F1、F2などの各ファンクション・キーに対応しています
また、ON、OFF、STOPでは、各キーごとに割り込みの状態を指定することが可能です。たとえば

```
KEY(1) ON  
KEY(2) OFF  
KEY(4) STOP
```

などです

2. ジョイスティックの発射ボタン(トリガーボタン)による割り込み
キーワード=STRIG

ジョイスティックのトリガーボタンまたは、本体のスペースキーを押すことに発生する割り込みです
ON STRIG GOSUBで指定する行番号は、それぞれのジョイスティックまたは、本体のスペースキーに対応しています
また、ファンクション・キーによる割り込みと同じく、ON、OFF、STOPでは、格ジョイスティックごとに指定が可能です。たとえば

```
STRIG(0) ON
```

```
STRIG(1) OFF
```

などです

3. ストップ・キーによる割り込み
キーワード=STOP

ストップ・キーが押されることによって発生する割り込みです

4. スプライトの衝突による割り込み
キーワード=SPRITE

スプライトの衝突によって発生する割り込みです。ゲームによく使用します

5. 時間による割り込み
キーワード=INTERVAL

ある決められた時間になると発生する割り込みです
ON INTERVAL zzzz GOSUBで飛び先を指定しますが、zzzzの値は60分の1秒Xzzzzとなります。たとえば
ON INTERVAL 120 GOSUB
の場合は、 $(1/60) \times 120 = 2$ 秒ごとに割り込みが発生することを示します

とかく BASIC で作ったアクションゲームは遅くてとても遊べない、などと言う人がいますが、この割り込み機能を十分使いこなせば、そうとうおもしろいプログラムを作ることができますよ。




ブロックくずし



ブロックくずし、と聞いて知らない人はいないと言えるほどポピュラーなゲームです。ようするにブロックと自分（パドル）の間にひとつのボールが飛びまわっていて、ブロックにボールがあたるとそのブロックが消えて得点になり、逆に自分の方にボールが来たときはパドルを動かして打ち返さなければならないわけです。

もし打ち返せなかったらボールが1つずつ減っていき、ゼロになったらゲームオーバーとなります。



遊び方

とにかくまず遊びたい人は、リスト1を注意深く入力して、カセットにセーブしておいてください。というのも、このプログラムは一部機械語を使用しており、1480行以後の機械語データの入力をミスしますと、あわれマイコンは暴走し、入力したプログラムはすべてパーになってしまいます。ちゃんとリストを見くらべて、正確に入力できたらRUNします。

STRING(J)

まずタイトルが出て、ジョイスティックで遊ぶか、MSX本体のカーソルキーで遊ぶか聞いてきますので、ジョイスティックの場合はY、カーソルキーの場合はNを入力してください。なお、ジョイスティックは、No.1側に接続してください。

これに答えると次にトリガーボタンを押すように指示が出ますので、本体キーの場合はスペースキー、ジョイスティックの場合はジョイスティックに付いているボタンを押すと、ボールが発射されゲームスタートとなります。

あとは必死になってボールを打ち返すのみです。例によって、ボ

ールが天井に当たるとパドルの長さが半分となり、ますます難しくなります。さらにボールがパドルのどの位置に当たったかによって、ボールの反射角およびスピードが変化しますので、かなり本物に近くなっています。BASICでやるには、この程度が限度でしょう。

得点はブロック1個につき10点で、何点取ってもボールの数は増えません。あしからず。ゲームオーバーになりますと、再ゲームするかどうかが聞いてきますので、再ゲームする場合はトリガーボタンを押してください。このときにハイスコアの更新も行われます。



プログラム

SCREEN 2

次にプログラムについて解析しましょう。まずこのプログラムを作成するとき一番問題になったのが、高解像度グラフィック (**SCREEN 2**) でのテキスト表示の遅さです。

このゲームでは、ブロックにボールが当たるとスコアの更新を行っているのですが、BASIC でこれを行うとあまりの遅さに、ボールが一瞬止ってしまうという現象が発生しました。

あれやこれやいろいろとやってみたのですが、どうも BASIC ではうまくゆかず、ついに機械語で表示することにしました。リストを見てもらえばわかるように、1480行から1700行までのデータが機械語プログラムで、プログラムの起動時に260~300行でメモリー内に機械語プログラムを作っています。

USR 0

USR 1

USR 2

DEFUSR

CLEAR

機械語プログラムはD000番地からD0AF番地までですが、内部で3つに分かれており、それぞれD000番地 (**USR 0**) が点数を0にするプログラム、D010番地 (**USR 1**) が点数に10点たして表示するプログラム、そしてD021番地 (**USR 2**) が現在の点数を表示するプログラムです。これら機械語プログラムの宣言 (**DEFUSR**) および機械語プログラム用のメモリーの確保 (**CLEAR**) は、100行、110行で行っています。このプログラムによって、ようやく遊ぶことのできるレベルになりました。

次に BASIC の部分についてお話ししましょう。

SPRITE\$ (0)

SPRITE\$ (1)

LINE

ボールおよびパドルは、スプライトを使用しています。このスプライトは、ゲームを作成するのに強力な命令で、任意の図形をスムーズにかつ高速に移動させることが可能です。スプライトの初期化は350~480行で行っており、**SPRITE\$ (0)** にボールの図形データ **SPRITE\$ (1)** にパドルの図形データを入れています。

ブロックは **LINE** 命令で枠を書き、ぬりつぶすことによって作成しています (620~690行)。ボールやパドルの移動は単にX軸方向、Y軸方向に移動量をたして行き、スプライトを表示すれば良く、ボールのスピードを変化させるには、この移動量を変化すれば良いわけです。このボールの移動量はXS, YS に、ボールの位置はXB, YB に、パドルは横方向にしか動けないのでY軸方向は固定で、X軸は

XP にそれぞれ入っています。

次が一番のノウハウである衝突の処理です。ボールは壁、パドル、ブロックの3つのものに衝突するわけですが、この処理が遅いとボールの動きがにぶくなり、ゲームがおもしろくなくなってしまいます。そこでこのプログラムでは、この3つのものと衝突した場合の判断をそれぞれ異なる方法で行い、高速化を実現しています。

ON~SPRITE~ GOTO (GOSUB)

まず壁との衝突ですが、これは壁の存在する座標が X, Y 軸共に分けて処理しやすいので、座標によって判断することにしました。950~990行あたりがこの処理を行っている部分です。次にパドルとの衝突ですが、スプライト同士の衝突については MSX では **ON SPRITE GOTO (GOSUB)** というとても便利な命令があるのでこれを使用しました。この命令は、いちいち2つのスプライトの座標を計算して衝突したかの判断をしなくても、2つ以上のスプライトが衝突すると自動的に指定した飛び先またはサブルーチンに分岐してくれるものです。330行、800行でこの制御を行っています。

POINT

最後にブロックとの衝突ですが、これは **POINT** 命令でそこがブロックの色かどうかを見て、ブロックとの衝突を判断しています。**POINT** 命令は、任意の座標の点の色を調べるための命令で、ボールの移動先は何もないか、ブロックかの2つの状態しかありませんので、これによって簡単に判断ができるわけです。

PAINT

なお、ブロックの消去は **PAINT** 命令で黒くぬりつぶすことによって行っています。

このプログラムは16Kバイトシステム以上で動作し、ディスクシステムでも使用できます。

フロッピーディスクをお使いの方へ

あなたがフロッピーディスクを使っているならば、プログラムの以下のリストのように一部訂正してください。これで、バッチリ。

```
100 MAXFILES=3: CLEAR 900, &HFFFF: DEFINT A-Z: DEFUSR=&HD000
110 DEFUSR1=&HD010: DEFUSR2=&HD021: OPEN "GRP:" FOR OUTPUT AS #1
260 AD=&HD000: RESTORE 1490
1490 DATA 21, 51, D0, 3E, 06, 36, 00, 23
1510 DATA 21, 52, D0, 06, 05, 7E, 3C, 27
1530 DATA F4, 21, 51, D0, 06, 06, 7E, 87
1540 DATA 87, 87, C5, E5, 21, 57, D0, 16
```

リスト I (その I) ブロックくずし

```

10 ' Block くずし
20 ' by Tamae Tama
30 ' debug and advice by Haruka takagi
40 ' copyright 1984 (C) NATS
50 ' copyright 1984 (C)
60 ' Seibundo-Shinkousha publishing
70 '
80 ' HENSU & SYSTEM INIT
90 '
100 MAXFILES=3: CLEAR 900, &HFFFF: DEFINT A-Z: DEFUSR=&HE00
    0
110 DEFUSR1=&HE010: DEFUSR2=&HE021: OPEN "GRP:" FOR OUTPUT
    AS #1
120 XS=0: YS=0: ' BALL SPEED X, Y
130 XB=0: YB=0: ' BALL X, Y
140 XP=0: ' PADL X
150 TP=0: ' HI- SCORE
160 SCREEN 1: CLS
170 PRINT "*****";
180 PRINT "*          Block くずし          *";
190 PRINT "*";
200 PRINT "*      by T. Tama & H. takagi      *";
210 PRINT "*      copyright 1984 (C)          *";
220 PRINT "*      Seibundo-Shinkousha          *";
230 PRINT "*****";
240 PRINT: PRINT
250 INPUT "      JOYSTICK ? (Y/N) "; I$
260 AD=&HE000: RESTORE 1490
270 READ D$: IF D$="end" THEN GOTO 310
280 D=VAL("&H"+D$): POKE AD, D
290 AD=AD+1
300 GOTO 270
310 J=0: IF I$="Y" OR I$="y" THEN J=1
320 SCREEN 2, 0, 0: KEY OFF
330 ON SPRITE GOSUB 1280
340 BALL=4: SCENE=-1: HIT=0: U=USR(0)

```



```
350 '
360 ' SPRITE INIT
370 '
380 RESTORE
390 SCREEN 2,0
400 S0$="":S1$=""
410 FOR I=0 TO 7
420 READ R
430 S0$=S0$+CHR$(R)
440 S1$=S1$+CHR$(255):NEXT
450 SPRITE$(0)=S0$
460 SPRITE$(1)=S1$
470 SCENE=SCENE+1
480 CL=(SCENE AND 3)*2+8
490 '
500 ' DRAW GAME STAGE
510 '
520 COLOR15,1,0:CLS:PRESET(8,24):PRINT#1,"BLOCK"
530 PRESET(8,32):PRINT#1,"ク スシ"
540 PRESET(8,56):PRINT#1,"SCORE"
550 PRESET(8,65):PRINT#1,"¥¥¥¥¥¥"
560 PRESET(8,96):PRINT#1,"TOP"
570 TP$=RIGHT$("0000"+RIGHT$(STR$(TP),LEN(STR$(TP))-1),
5)
580 PRESET(0,104):PRINT#1,USING" & &";TP$
590 U=USR2(0)
600 PRESET(8,152):PRINT#1,"BALL"
610 LINE(47,0)-(240,192),15,B
620 LINE -(47,192),0
630 LINE(48,20)-(238,60),CL,BF
640 FOR I=20 TO 60 STEP 8
650 LINE(48,I)-(239,I),1
660 NEXT
670 FOR I=48 TO 239 STEP 12
680 LINE(I,20)-(I,60),1
690 NEXT
700 XP=100:HIT=0
710 GOSUB1210
720 XS=1:YS=-4:XB=100:YB=100
730 PUT SPRITE 1,(XP,185),12,1:PUT SPRITE2,(XP+8,185),1
2,1
```

```

740 '
750 ' START ?
760 '
770 PSET (70,100),0:PRINT #1,"トリカ ホ`タンヲ オシテクタ`サイ":F=0
780 IF STRIG(J)=0 THEN 780
790 LINE(70,100)-(230,108),0,BF
800 BEEP:SPRITE ON
810 GOSUB 1210
820 '
830 ' PADL MOVE
840 '
850 FOR HIT=HIT TO 79
860 IF 3=STICK(J) AND 223+8*F>XP THEN XP=4+XP ELSE 890
870 PUT SPRITE2,(8+XP,185+24*F),12,1
880 PUT SPRITE1,(XP,185),12,1:GOTO 950
890 IF 7=STICK(J) AND 48<XP THEN XP=XP-4 ELSE 950
900 PUT SPRITE1,(XP,185),12,1
910 PUT SPRITE2,(8+XP,185+24*F),12,1
920 '
930 ' BALL MOVE
940 '
950 XB=XB+XS:YB=YB+YS
960 IF 231<XB THEN XS=-ABS(XS):PLAY"L64A":GOTO 980
970 IF 48>XB THEN XS=ABS(XS):PLAY"L64A"
980 IF 1>YB THEN YS=ABS(YS):PLAY"L64A":Y=5:F=1:PUTSPRIT
    E 2,(0,209),0,0
990 IF 200<YB THEN 1100
1000 PUT SPRITE0,(XB,YB),12,0
1010 '
1020 ' HIT ?
1030 '
1040 IF CL-POINT(3+XB,3+YB) THEN 860
1050 PLAY"L64B":PAINT(3+XB,3+YB),1
1060 YS=-YS:U=USR1(0):NEXT:GOTO 380

```

```
1070 '
1080 ' BALL OUT
1090 '
1100 PLAY"L501A":BALL=BALL-1
1110 FOR I=0 TO 1000:NEXT
1120 IF BALL>0 THEN 720
1130 PRESET(80,80):PRINT #1,"GAME OVER":SC=(HIT+SCENE*8
0)*10
1140 IF TP<SC THEN TP=SC
1150 PRESET(80,96):PRINT #1,"REPLAY : トリカ` ホ`タン"
1160 IF STRIG(J)=0 THEN 1160
1170 GOTO 340
1180 '
1190 ' NOKORI BALL WRITE
1200 '
1210 C=12:FOR I=1 TO 5
1220 IF I=BALL THEN C=1
1230 PUT SPRITE I+2, (8*I,160),C,0
1240 NEXT:RETURN
1250 '
1260 ' SPRITE WARIKOMI
1270 '
1280 SPRITE OFF:BEEP
1290 ON 3+F+(XB-XP)/4 GOTO 1300,1310,1340,1340,1330,132
0
1300 XS=-5:GOTO 1340
1310 XS=-2:GOTO 1340
1320 XS=5:GOTO 1340
1330 XS=2
1340 YS=-ABS(YS):XB=XB+XS:YB=YB+YS
1350 PUT SPRITE 0, (XB,YB),12,0
1360 SPRITE ON:RETURN
```

リスト I (その 5) ブロックくずし

```

1370 '
1380 ' BALL SPRITE DATA
1390 '
1400 DATA &B00111100
1410 DATA &B01111110
1420 DATA &B11111111
1430 DATA &B11111101
1440 DATA &B11111111
1450 DATA &B11111001
1460 DATA &B01110010
1470 DATA &B00111100
1480 ' machine language data
1490 DATA 21,51,E0,3E,06,36,00,23
1500 DATA 3D,20,FA,C9,00,FE,41,20
1510 DATA 21,52,E0,06,05,7E,3C,27
1520 DATA E6,0F,77,20,04,23,05,20
1530 DATA F4,21,51,E0,06,06,7E,87
1540 DATA 87,87,C5,E5,21,57,E0,16
1550 DATA 00,5F,19,E5,21,F8,07,CB
1560 DATA 20,CB,20,CB,20,48,06,00
1570 DATA 09,54,5D,E1,01,08,00,CD
1580 DATA 5C,00,E1,C1,23,05,20,D6
1590 DATA C9,00,00,00,00,00,00,00
1600 DATA 3C,46,46,46,46,46,3C,00
1610 DATA 1C,3C,7C,1C,1C,1C,7E,00
1620 DATA 3C,46,46,1C,30,60,7E,00
1630 DATA 3C,46,46,0C,46,46,3C,00
1640 DATA 06,0E,16,26,7E,06,06,00
1650 DATA 7C,60,60,7C,06,46,3C,00
1660 DATA 1C,20,40,7C,46,46,3C,00
1670 DATA 7E,46,46,0C,18,18,18,00
1680 DATA 3C,46,46,3C,46,46,3C,00
1690 DATA 3C,46,46,3E,06,06,3C,06
1700 DATA 00,FF,00,FF,00,FF,end


```

ホームデータ ベース



みなさん、自分のもっているプログラム（カセットテープに入っているもの）やレコード、友だちの住所などを整理するとき、どのように行っていますか？ もちろん、ノートに付けて行く方法もありますが、それでは現在の情報化時代に取り遅れてしまいますし、せっかくマイコンがあるんですから、これにやらせてみようではありませんか。

このプログラムは、言うならばコンピューターによるメモ帳で、あらかじめ入れておいたデータ（たとえば、レコードの名前、値段、買った日、保存場所）に対して、キーワード（探すのにカギとなる文字）を入力することにより、目的の情報をすばやく探し出すことが可能なものです。



使い方

まずリスト2を正確に入力してください。このプログラムはずいぶんとつまっており、ごちゃごちゃして見にくいのですが、ゆったりと見やすくプログラムを作ると、プログラムの大きさが大きくなってしまい、かんじんのデータ領域が小さくなってしまうので泣く泣く、こうなってしまったわけです。


RUN

リスト2を入力したら、**RUN** とすると一定時間タイトルが表示され、メニューが表示されます。次にそれぞれのモードについて解説しましょう。

1. キーボードからのデータ入力

1を押すとキーボードからのデータ入力モードとなります。各データは6つの項目を持っており、それぞれ何を入れるかは自分で決めることができます。また、内容としては文字、数字が使用できますが、特殊文字として「, (カンマ)」および「\ (ダブルコーション)」は使用できませんので注意してください。

データの数100個まで使用できるように配列を宣言していますが、RAMの容量や1つデータの文字数によって、100件では多すぎてエラーとなる場合や、100件入れてもメモリー上に余裕ができてしまう場合がありますので、これは自分で調整してみてください。なお、データは追加されていく形で入りますので、前のデータが消されてしまう、といった心配はありません。



2. テープからのデータ入力

まず、キーボードから入力して作ったデータを3のテープへの出力を使って、カセットテープに出力しておきます。そして、この次以後は、いちいちデータをもう一度入力しなくても、この命令によりカセットテープからデータを入力することができるわけです。

また、2つの別々に作っておいたデータを、この命令で読み込むことによって、1つにすることが可能です（アペンド機能）。

3. テープへのデータ出力

現在メモリー上にあるデータをカセットテープにすべて出力します。逆にテープからデータを読み込むのは2の命令を使用します。

4. データの削除

データの入力のときにミスをして入れてしまったデータや不要になったデータを消すときに使用します。

データの消しかたは、キーワードを入力することにより、次々と該当するデータを表示しますので、消したいデータが見つかったらYを押しますと、データを消すことができます。

5. データの検索

まず、検索するデータの各項目のキーワードを聞いてきますので入力して行きます。どうしてもよい項目については、リターンキーのみ入力すれば次の項目に進んでくれます。キーワードの入力を終了すると、この条件に基き、データの中から該当するものを次々と捜し出して表示を行います。なお、このプログラムでは画面のみでプリンターなどには出力できませんが、840行の **PRINT** 文を **LPRINT** 文に変更することによってプリンターにも出力することが可能です。

6. 全データの表示

現在メモリー内に存在する全データの表示を行います。つまり、データの検索で、キーワードをすべてリターンキーのみしか押さなかったときと同じ働きをするわけです。

7. 終了

このプログラムの実行を終了します。ここでは、エラートラップの解除 (**ON ERROR GOTO 0**) を行っていますので、必ずこの命令によってプログラムを終了させ、ストップキーでは止めないようにしてください。

PRINT
LPRINT

ON ERROR
GOTO 0

プログラム

プログラムの処理の中心は、配列Kに入っているデータの書き換え（入力）、交換（削除）および表示が主なもので、現在のデータの個数は変数Pに入っています。

SWAP

特に複雑な処理を行っている部分はありませんが、データの削除方法として **SWAP** 命令を使用した変数の入れ替え方式を使用している点が、少し高度な処理に入るほうでしょうか。

ON J GOTO

各処理は、330行の **ON J GOTO** 文によって飛んでおり、

350行=キーからのデータ入力

510行=テープからのデータ入力

570行=テープへのデータ出力

630行=データの削除

750行=データの検索

830行=全データの表示

880行=終了

となっています。

このプログラムは16Kバイト以上のシステムで使用可能で、ディ

OPEN "CAS : DATA"

OPEN "DATA"

スクシステムでも使用可能です。なお、ディスクシステムでは、**OPEN "CAS : DATA"** の部分を **OPEN "DATA"** にするとテープではなく、ディスクに対してデータの読み書きを行いますので、テープよりずっと速く処理が行えます。



```

10 ' Home data-base program
20 ' original by Haruka Takagi
30 ' by Mika Tada
40 ' by Haruka Takagi
50 ' copyright 1984 NATS
60 ' copyright 1984 (C)
70 '      Seibundo-Shinkousha publishing
80 DEFINT A-Z:P=0
90 SCREEN 1:CLS:COLOR 15,0,0:KEY OFF
100 ON ERROR GOTO 930
110 PRINT "*****";
120 PRINT "*"      Home Data-base program  "*";
130 PRINT "*"
140 PRINT "*"      by M.Tada & H.takagi  "*";
150 PRINT "*"      copyright 1984 NATS    "*";
160 PRINT "*"      copyright 1984 (C)     "*";
170 PRINT "*"      Seibundo-Shinkousha   "*";
180 PRINT "*****";
190 FOR TM=0 TO 5000:NEXT
200 DIM D$(5,100)
210 CLS:PRINT "1...キー から にゅうりょく"
220 PRINT "2...テーブル から にゅうりょく"
230 PRINT "3...テーブル に しゅうりょく"
240 PRINT "4...さくし"
250 PRINT "5...けんさく"
260 PRINT "6...ひょうし"
270 PRINT "7...おしまい"
280 PRINT :PRINT "しごと の は"んご"う は (1-7)";
290 J$=INPUT$(1):J=VAL(J$)
300 IF J>7 OR J<1 THEN 290
310 PRINT J
320 PRINT
330 ON J GOTO 350,510,570,630,750,830,880

```

```

340 'data input
350 P=P+1:PRINT
360 FOR I=0 TO 5
370 PRINT USING "テータ (#)";I+1;:INPUT D$(I,P):NEXT:PRINT
380 FOR I=0 TO 5:PRINT USING "テータ (#) ";I+1;:PRINT D$(I,P):NEXT:PRINT
390 PRINT "これで いいですか ";
400 INPUT A$:IF A$="N" OR A$="n" THEN 410 ELSE 480
410 PRINT "なんはんの テータ を なおしますか ( 1-6 ) ";
420 INPUT N
430 IF N>6 OR N<0 THEN 420
440 PRINT USING "テータ (#) ";N;:PRINT D$(N-1,P);"   これで
    "すか";
450 INPUT A$
460 IF A$="Y" OR A$="y" THEN 470 ELSE IF A$="N" OR A$="n" THEN 410 ELSE 450
470 PRINT USING "テータ (#)";N;:INPUT D$(N-1,P):PRINT :GO
    TO 380
480 INPUT "にゅうりよく を つづけますか";A$:IF A$="n" OR A$="N" THE
    N GOTO 210 ELSE GOTO 350
    
```





```
490 'read from tape
500 '
510 PRINT :INPUT "カセットテープの しゅんぴ は いいてすか ";A$:IF A$="
    N" OR A$="n" THEN 210
520 OPEN "CAS:DATA" FOR INPUT AS#1
530 INPUT #1,PA:IF PA+P>100 THEN CLOSE #1:PRINT :PRINT
    "テープか" おおすきます !!":GOTO 210
540 FOR L1=1 TO PA:FOR L2=0 TO 5:INPUT #1,D$(L2,P+L1):N
    EXT :NEXT :CLOSE #1:P=P+PA
550 PRINT "テープ にゅうりょく かんりょう":PRINT:GOTO 850
560 'write to tape
570 IF P=0 THEN PRINT :PRINT "てーたか" ありません !! ":PRINT :
    GOTO 850
580 PRINT :INPUT "カセットテープの しゅんぴ は いいてすか ";A$:IF A$="
    N" OR A$="n" THEN 210
590 OPEN "CAS:DATA" FOR OUTPUT AS#1:PRINT #1,P
600 FOR L1=1 TO P:FOR L2=0 TO 5:PRINT #1,D$(L2,L1):NEXT
    :NEXT :CLOSE #1
610 PRINT :PRINT "テープを テープに しつりょく しました !!":PRINT :BEE
    P:GOTO 850
620 'delete data
630 IF P=0 THEN 830
640 KS=1
650 GOSUB 890
660 GOSUB 900
670 IF KL=1 THEN 700
680 KS=KS+1:IF KS=<P THEN 660
690 BEEP:PRINT :PRINT "テープか" なくなりました !!":PRINT :GOTO 8
    50
700 PRINT :INPUT "この テープを けしますか ";A$:IF A$="Y" OR A$="
    y" THEN 710 ELSE 680
710 FOR L1=KS TO P-1:FOR L2=0 TO 5:SWAP D$(L2,L1+1),D$(
    L2,L1):NEXT :NEXT
720 P=P-1
730 PRINT :PRINT "テープを けしました !! ":PRINT :GOTO 850
```

```

740 'search data
750 IF P=0 THEN 830
760 KS=1
770 GOSUB 890
780 GOSUB 900
790 KS=KS+1:IF KS>P THEN 690
800 IF KL=0 THEN 780
810 GOTO 780
820 'display all data
830 IF P=0 THEN PRINT :PRINT :PRINT "また" データ か" ありません
!! ":BEEP:PRINT :GOTO 850
840 PRINT :FOR L1=1 TO P:FOR L2=0 TO 5:PRINT USING "データ
( #)";L2+1;:PRINT D$(L2,L1):NEXT :FOR W=0 TO 500:N
EXT W:PRINT :NEXT
850 PRINT "つき"の しよりを するために なにか キー を おして くだ"さい !"
860 IF INKEY$="" THEN 860 ELSE 210
870 'end of job
880 PRINT :PRINT :PRINT "こ"くろうさまで"した":ON ERROR GOTO 0:EN
D
890 PRINT :FOR L=0 TO 5:PRINT USING "データ ( #) キーワード ";L
+1;:KW$(L)="":INPUT KW$(L):NEXT:RETURN
900 KA=1:FOR L1=0 TO 5:KK=INSTR(D$(L1,KS),KW$(L1)):KA=K
A*KK:NEXT
910 IF KA=0 THEN KL=0:RETURN
920 PRINT :PRINT :FOR L1=0 TO 5:PRINT USING "データ ( #)
";L1+1;:PRINT D$(L1,KS):NEXT :KL=1:RETURN
930 CLS
940 PRINT "エラー か" はっせい しました!!!"
950 PRINT "もういちど" やりなおしてくだ"さい"
960 INPUT "なにか キーを おしてくだ"さい"
970 RESUME 210

```

スーパーバックアップ プログラム



みなさんの周りにも、MSX マシンを持っている人が多いことと思いますが、ちょっと人の持っているカセットテープのプログラムをもらいたいな、と思ったときも、一本にたくさんはっているとなかなかめんどうですね。

こんなときには、いまはやりのダブルデッキなんかを使って移すこともできるのですが、パソコンのプログラムやデータの信号（音）というものは、実に微妙で、ちょっとドロップアウトがあったり、波形がなまっていたりすると、すぐにエラーとなってしまいます。

ほんとでしたら、プログラムひとつひとつを本体に読み込んでからセーブするのがベストなのですが、これは手間がかかり、時間もかかるのでやりたくないですね。でも、確実に移しておきたいなんて、ちょうしの良いことができないでしょうか。

このプログラムは、これを可能にしたもので、使い方は実に簡単。ダブルデッキで移すのと同じ要領でうまくバックアップを使うことができます。

```
Cassette copy program for MSX
Version 1.1
by Haruka Takagi
copyright 1984 NATS

PHASE invert (y/n)n
SOUND output (y/n)y
```

プログラムの原理

MSX がテープにプログラムやデータをセーブするときは、メモリーの「1」と「0」を音に変えてセーブするわけですが、この音は図 1a のような形をしています。ちょうど正弦波（わからない人は、わからなくてもけっこうです）が少しゆがんだような形ですね。

さてこの信号が、MSX のカセット読み込み回路に入ると、図 1b のように方形の波形に変えられます。つまり、「0」と「1」に波形が変形されたと考えられます。そこで、この波形をすごく短い時間で切って、その時の「1」か「0」かを逆にカセットの出力回路に送ってやれば、ほぼ似たような波形が得られることになります（図 1c 参照）。

まず、2 台のカセットテレコを用意し、片方を再生しその信号を MSX に入力。このプログラムによってきれいになった信号が再びカセット接続ケーブルにもどって、もう一台のテレコに出力して記録すれば、ダブルデッキよりも良好なバックアップを作ることができます。

ただし、バックアップをくり返すと、テレコのワウ・フラッターの影響が大きくなりエラーが発生しますので、この方法でのバックアップは 1 回が限度と考えるのが良いでしょう。それに個人で使用する場合、バックアップは 1 本あればまずだいじょうぶですからね。

2・4 kHz の方形波を入力し（下）、出力（赤ブ
ラグ）よりカセットテープに記録する波形が上。





図 1



プログラム

DATA

リスト 3a にスーパーバックアッププログラムのリストを示します。一見すべて BASIC で書かれているように見えますが、実は、BASIC では単にメモリー上に機械語プログラムを作っているだけなのです。機械語のプログラムは220行以後の **DATA** 文で構成されています。

なぜ機械語を使うかと言うと、先ほど話しましたが、カセットから入力される波形をすごく短い時間で切るには、BASIC では遅すぎるためなのです。また、機械語そのものは人間にわかりにくい(ただの数字です)ので、アセンブラと呼ばれる言語を使用し、これを機械語に変換するわけです。このプログラムをアセンブラで書いたものをリスト 3b に示します。

ちんぷんかんぷんの人も多いと思いますが、MSX をほんとうに使いこなすためには、必ずアセンブラが必要となります。みなさんがアセンブラを勉強し始めたら、必ずこのリストが役に立ちますので、それまでこの本はだいに保存してくださいね。

プログラムの使い方

まず、リスト3aを入力し、必ずセーブしておきます。このプログラムは機械語を使用しているため、**DATA** 文のデータを1つでも間違えると暴走するからです。


次に、図2のようにMSXと2台のテレコを継ぎます。2台のテレコは、できれば同じものがベストです。そして、再生用テレコにマスターテープを、記録用テレコに新しいテープを入れ、このプログラムをRUNします。

タイトルが出てしばらくすると、**PHASE invert (y/n)**と聞いてきますので、2台のテレコの位相関係が逆の場合はy、同じ場合はnを入れてください。わからない場合は、バックアップしたテープがエラーを起したりする場合、yかnを適当に入れかえてちゃんと読めるほうにしてください。

次に**SOUND output (y/n)**と聞いてきますので、カセットの音をMSXのオーディオ出力端子から出したい場合はy、出たくない場合はnを入力してください。

これで準備が完了しました。あとは再生用テレコを再生し、記録用テレコを記録(録音)にすれば自動的にカセット1本分をバックアップしてくれます。なお、このプログラムは一度動き出すと電源を切る以外に止めることはできません。また、このプログラムは16Kバイト以上のシステムで使用可能で、ディスクがつながっていてもOKです。

図2



```

10 ' Super cassette back up program
20 ' by Haruka Takagi
30 ' copyright 1984 (C) NATS
40 ' copyright 1984 (C)
50 ' Seibundo-Shinkousha publishing
60 MAXFILES=0: CLEAR 10, &HF1FF
70 SCREEN 0: CLS
80 PRINT "*****";
90 PRINT "*      Super cassette back up program      *";
100 PRINT "*      by Haruka Takagi                        *";
110 PRINT "*      copyright 1984 (C) NATS                  *";
120 PRINT "*      copyright 1984 (C)                      *";
130 PRINT "*      Seibundo-Shinkousha publishing          *";
140 PRINT "*****";
150 PRINT: PRINT: PRINT
160 PRINT "Please wait for a moment!!!"
170 AD=&HF200
180 READ D$: IF D$="end" THEN GOTO 220
190 D=VAL("&H"+D$)
200 POKE AD, D: AD=AD+1
210 GOTO 180
220 DEF USR=&HF200
230 A=USR(0)

```




—リスト 3a(その2) スーパーバックアップ—

```

240 'machine language data
250 '&hF200 -> &hF300
260 '
270 DATA AF,32,FE,F2,11,7F,F2,CD
280 DATA 76,F2,11,D5,F2,CD,76,F2
290 DATA CD,9F,00,F5,CD,A2,00,11
300 DATA FB,F2,CD,76,F2,F1,FE,6E
310 DATA 28,0E,FE,79,20,E4,3E,2F
320 DATA 32,5D,F2,3E,80,32,FE,F2
330 DATA 11,E8,F2,CD,76,F2,CD,9F
340 DATA 00,F5,CD,A2,00,11,FB,F2
350 DATA CD,76,F2,F1,FE,6E,20,07
360 DATA 3E,20,32,65,F2,18,04,FE
370 DATA 79,20,DD,21,FE,F2,F3,3E
380 DATA 0E,D3,A0,DB,A2,00,E6,80
390 DATA BE,28,F4,A7,3E,A0,28,07
400 DATA D3,AA,3E,80,77,18,E8,3E
410 DATA 00,D3,AA,77,18,E1,1A,A7
420 DATA C8,CD,A2,00,13,18,F7,0C
430 DATA 43,61,73,73,65,74,74,65
440 DATA 20,63,6F,70,79,20,70,72
450 DATA 6F,67,72,61,6D,20,66,6F
460 DATA 72,20,4D,53,58,0D,0A,56
470 DATA 65,72,73,69,6F,6E,20,5D
480 DATA 5B,0D,0A,62,79,20,48,61
490 DATA 72,75,6B,61,20,54,61,6B
500 DATA 61,67,69,0D,0A,63,6F,70
510 DATA 79,72,69,67,68,74,20,31
520 DATA 39,38,34,20,4E,41,54,53
530 DATA 0D,0A,0D,0A,00,50,48,41
540 DATA 53,45,20,69,6E,76,65,72
550 DATA 74,20,28,79,2F,6E,29,00
560 DATA 53,4F,55,4E,44,20,6F,75
570 DATA 74,70,75,74,20,28,79,2F
580 DATA 6E,29,00,0D,0A,00,00,FF
590 DATA end

```


コンピュレコーダー バージョン2



CD (コンパクトディスク) や PCM などデジタルオーディオを聞いたことのある人も多いと思いますが、MSX を使って、このデジタルオーディオに挑戦してみたのが、このコンピュレコーダーです。

いつもは、プログラムやデータを読むのに使用しているカセット入力(白いプラグ) から、「音」を MSX のメモリーに読み込み、オーディオ出力端子から出そうと言うわけです。

いくらデジタルオーディオと言っても、A/D や D/A コンバーターのない MSX ですから、その音質は目をつぶることにして、一つの実験として楽しんでみてください。

さて、音という周波数の高い信号を扱うには、遅い BASIC ではとうてい不可能です。そこで機械語を使用するわけですが、みなさんの中には、機械語を扱ったことがない人も多いと思い、プログラムは一見 BASIC で作ってあるように見えるように開発しました (リスト 4 a)。

しかし、その実体はリスト 4 b に示すようなアセンブラと呼ばれるもので作られているのです。このアセンブラは、機械語のプログラムを開発するためのもので、みなさんはなじみがない人がほとんどだと思いますが、そのうち必ず役に立ちますのでよく覚えておいてください。

プログラムの使い方

DATA

まずリスト 4a を入力して必ずカセットにセーブしておきます。
機械語のプログラムですので、**DATA** 文のデータを 1 文字でも間違
いますと CPU が暴走するためです。


RUN

次に **RUN** しますと、メッセージが出て入力待ちとなります。以下
に各コマンドの説明をします（なおコマンドは、すべて小文字で入
力してください）。

1. t (テスト・コマンド)

t を押すと、カセット入力からの音を MSX 内で変換したものを
そのままオーディオ出力端子から出します。テレコの頭出しや入力
レベルの調整に使用してください。

このコマンドから抜けるには、コントロールスイッチと、ストッ
プスイッチを同時に押します。



2. r (録音コマンド)

r を押すと、まず記録スピードを聞いてきます。1 が最も速く、
9 が一番遅くなります。1 の場合
約 20 秒間、9 の場合は約 30 秒間 (32
K バイトシステムの場合) 記録す
ることができます。数字が大きい
ほうが記録時間は長いのですが、
逆に音質は悪くなります。

スピードを入力した後、何でも
良いからキーを押すと記録がスタ
ートします。メモリーいっぱい
に音を記録すると自動的にメニュー
に戻ります。

3. p (再生コマンド)

p を押すと再生スピードを聞いて
きます。1 が一番速く 9 が一番
遅くなります。r コマンドで記録
したスピードより速くすると早回

しに、逆に遅くすると、モゴモゴした感じになります。

再生をやめるには、コントロールスイッチとストップスイッチを同時に押します。

4. o (ループプレイコマンド)

このスイッチを押すたびに画面には、LOOP PLAYとNON-LOOP PLAYが交互に表示されます。

ループプレイとは、ちょうどエンドレスのテープと同じで、Pコマンド再生すると何度でも再生します。逆にノンループプレイは、1度再生したらメニューに戻ります。

5. s (セーブカセット)

sを押すと、現在メモリー上にある音のデータを、カセットにセーブします。32Kバイトシステムでは15分ぐらいかかりますので、お茶でも飲んで待つと良いでしょう。

6. l (ロードカセット)

l (エル)を入力すると、カセットから音のデータを読み込みます。セーブの場合と同じ時間がかかります。

このプログラムは16Kバイト以上のシステムで動作し、ディスクが見つないである場合ははずしてからプログラムを実行してください。



```
10 ' Compu recoder version )(
20 ' by Haruka Takagi
30 ' copyright 1984 (C) NATS
40 MAXFILES=0: CLEAR 10, &HF05F
50 SCREEN 0: CLS
60 PRINT "*****";
70 PRINT "*      Compu recoder version )(      *";
80 PRINT "*      by Haruka Takagi      *";
90 PRINT "*      copyright 1984 (C) NATS      *";
100 PRINT "*      copyright 1984 (C)      *";
110 PRINT "*      Seibundo-Shinkousha publishing      *";
120 PRINT "*****";
130 PRINT: PRINT: PRINT
140 PRINT "Please wait for a moment!!!"
150 AD=&HF060
160 READ D$: IF D$="end" THEN GOTO 200
170 D=VAL("&H"+D$)
180 POKE AD, D: AD=AD+1
190 GOTO 160
200 DEF USR=&HF060
210 A=USR(0)
220 'machine language data
230 '&hF060 -> &hF37f
```

```

240 '
250 DATA FB, 11, D5, F1, CD, 96, F0, 3A
260 DATA 7A, F3, A7, 11, 45, F3, 20, 03
270 DATA 11, 60, F3, CD, 96, F0, CD, 9F
280 DATA 00, FE, 72, 28, 3A, FE, 70, 28
290 DATA 7C, FE, 74, 28, 1A, FE, 73, CA
300 DATA 41, F1, FE, 6C, CA, 71, F1, FE
310 DATA 6F, CA, C8, F1, 18, CA, 1A, A7
320 DATA C8, CD, A2, 00, 13, 18, F7, 11
330 DATA 70, F2, CD, 96, F0, F3, 3E, 0E
340 DATA CD, 96, 00, E6, 80, CD, 35, 01
350 DATA CD, B7, 00, 38, AB, 18, EF, 11
360 DATA 8D, F2, CD, 96, F0, CD, B4, F1
370 DATA 32, 79, F3, 11, AF, F2, CD, 96
380 DATA F0, CD, 9F, 00, F3, 11, 60, F0
390 DATA 2A, 48, FC, 06, 08, 00, 3E, 0E
400 DATA CD, 96, 00, E6, 80, F5, CD, 35
410 DATA 01, F1, 17, CB, 16, CD, AB, F1
420 DATA 10, EC, CD, B7, 00, DA, 60, F0
430 DATA 23, E5, AF, ED, 52, 7C, B5, E1
440 DATA 20, D9, C3, 60, F0, 11, 7C, F2
450 DATA CD, 96, F0, CD, B4, F1, 32, 79
460 DATA F3, F3, 11, 60, F0, 2A, 48, FC
470 DATA 06, 08, 4E, 3E, 0E, CD, 96, 00
480 DATA E6, 00, CB, 11, 1F, F5, CD, 35
490 DATA 01, F1, CD, AB, F1, 10, EC, CD
500 DATA B7, 00, DA, 60, F0, 23, E5, AF
510 DATA ED, 52, 7C, B5, E1, C2, 10, F1
520 DATA 3A, 7A, F3, A7, 28, CC, C3, 60
530 DATA F0, 11, 98, F2, CD, 96, F0, CD
540 DATA 9F, 00, 2A, 48, FC, E5, 3E, FF
550 DATA CD, EA, 00, E1, 38, 49, 7E, E5
560 DATA CD, ED, 00, E1, 38, 41, 23, E5
570 DATA AF, 11, 60, F0, ED, 52, 7C, B5
580 DATA E1, 20, EB, CD, F0, 00, C3, 60
590 DATA F0, 11, BD, F2, CD, 96, F0, CD
600 DATA 9F, 00, 2A, 48, FC, E5, CD, E1
    
```

610 DATA 00,E1,38,1B,E5,CD,E4,00
620 DATA E1,38,14,77,23,E5,AF,11
630 DATA 60,F0,ED,52,7C,B5,E1,20
640 DATA EB,CD,E7,00,C3,60,F0,11
650 DATA E4,F2,CD,96,F0,CD,9F,00
660 DATA C3,60,F0,F5,3A,79,F3,3D
670 DATA 20,FD,F1,C9,11,14,F3,CD
680 DATA 96,F0,CD,9F,00,FE,31,38
690 DATA F3,FE,3A,30,EF,D6,30,C9
700 DATA 3A,7A,F3,E6,01,EE,01,32
710 DATA 7A,F3,C3,60,F0,0C,43,6F
720 DATA 6D,70,75,2D,72,65,63,6F
730 DATA 64,65,72,0D,0A,56,65,72
740 DATA 73,69,6F,6E,20,5D,5B,0D
750 DATA 0A,62,79,20,48,61,72,75
760 DATA 6B,61,20,54,61,6B,61,67
770 DATA 69,0D,0A,63,6F,70,79,72
780 DATA 69,67,68,74,20,31,39,38
790 DATA 34,20,4E,41,54,53,0D,0A
800 DATA 0D,0A,72,2E,2E,52,45,43
810 DATA 20,6D,6F,64,65,0D,0A,70
820 DATA 2E,2E,50,42,20,20,6D,6F
830 DATA 64,65,0D,0A,74,2E,2E,54
840 DATA 45,53,54,20,6D,6F,64,65
850 DATA 0D,0A,73,2E,2E,53,41,56
860 DATA 45,20,64,61,74,61,0D,0A
870 DATA 6C,2E,2E,4C,4F,41,44,20
880 DATA 64,61,74,61,0D,0A,6F,2E
890 DATA 2E,4C,4F,4F,50,20,70,6C
900 DATA 61,79,20,73,77,0D,0A,00
910 DATA 54,65,73,74,20,6D,6F,64
920 DATA 65,0D,0A,00,50,4C,41,59
930 DATA 20,42,41,43,4B,20,4D,4F
940 DATA 44,45,0D,0A,00,52,45,43
950 DATA 20,4D,4F,44,45,0D,0A,00
960 DATA 53,41,56,45,20,54,4F,20
970 DATA 43,41,53,53,45,54,54,45
980 DATA 20,54,41,50,45,0D,0A,48
990 DATA 49,54,20,41,4E,59,20,4B

—リスト4a(その4) コンピュレコーダーⅡ—

```
1000 DATA 45, 59, 0D, 0A, 00, 4C, 4F, 41
1010 DATA 44, 20, 46, 52, 4F, 4D, 20, 43
1020 DATA 41, 53, 53, 45, 54, 54, 45, 20
1030 DATA 54, 41, 50, 45, 0D, 0A, 48, 49
1040 DATA 54, 20, 41, 4E, 59, 20, 4B, 45
1050 DATA 59, 0D, 0A, 00, 43, 41, 53, 53
1060 DATA 45, 54, 54, 45, 20, 49, 53, 20
1070 DATA 45, 52, 52, 4F, 52, 20, 4F, 52
1080 DATA 20, 41, 42, 4F, 52, 54, 45, 44
1090 DATA 20, 21, 21, 07, 0D, 0A, 48, 49
1100 DATA 54, 20, 41, 4E, 59, 20, 4B, 45
1110 DATA 59, 0D, 0A, 00, 49, 4E, 50, 55
1120 DATA 54, 20, 52, 45, 43, 2F, 50, 42
1130 DATA 20, 53, 50, 45, 45, 44, 20, 28
1140 DATA 31, 7E, 39, 29, 0D, 0A, 31, 2E
1150 DATA 2E, 2E, 46, 41, 53, 54, 0D, 0A
1160 DATA 39, 2E, 2E, 2E, 53, 4C, 4F, 57
1170 DATA 0D, 0A, 0D, 0A, 00, 28, 6E, 6F
1180 DATA 6E, 2D, 6C, 6F, 6F, 70, 20, 70
1190 DATA 6C, 61, 79, 20, 6D, 6F, 64, 65
1200 DATA 20, 6E, 6F, 77, 29, 0D, 0A, 00
1210 DATA 28, 6C, 6F, 6F, 70, 20, 70, 6C
1220 DATA 61, 79, 20, 6D, 6F, 64, 65, 20
1230 DATA 6E, 6F, 77, 29, 0D, 0A, 0D, 0A
1240 DATA 00, 00, 00, FF, 00, 00, FF, 04
1250 DATA end
```


リスト4b(その3)

2000i	F10A 1160F0	PLP:	LD	DE,RTOP
2010i	F10D 2A4BFC		LD	HL,(BOTTOM)
2020i	F110 0A60B	FB2:	LD	B,B
2030i	F112 4E		LD	C,(HL)
2040i	F113 0A60B	FB1:	CALL	DE,RTOP
2050i	F113 CD9A60		CALL	DE,RTOP
2060i	F118 E6A0		AND	00H
2070i	F11A CB11		RL	C
2080i	F11C 1F		RRR	
2090i	F11D FS		PUSH	AF
2100i	F11E CD3501		CALL	CHSND
2110i	F11F 0A60B		CALL	DE,RTOP
2120i	F122 CDABF1		CALL	DE,RTOP
2130i	F125 10EC		DJNZ	PB1
2140i	F127 CD9700		CALL	BREAKX
2150i	F12A DA6A0F0		CALL	C,RTOP
2160i	F12D 23		INC	HL
2170i	F12E 77		INC	HL
2180i	F12E AF		XOR	HL
2190i	F130 ED52		SBC	HL,DE
2200i	F132 7C		LD	A,H
2210i	F133 B5		OR	L
2220i	F13A E1		POP	HL
2230i	F13B 210F1		JP	NZ,PB2
2240i	F13C 3A7AF3		JP	NZ,PB2
2250i	F13B A7		AND	A,(FLAG)
2260i	F13C 28CC		JP	Z,PLP
2270i	F13E C3A0F0		JP	RTOP
2280i	F141 1198F2	SAVE:	LD	DE,SVMSG
2290i	F142 CD9A60		CALL	DE,RTOP
2300i	F147 CD9A60		CALL	DE,RTOP
2310i	F14A 2A4BFC		LD	HL,(BOTTOM)
2320i	F14D E5		PUSH	HL,A,OFFH
2330i	F14E 3EFF		LD	A,OFFH
2340i	F150 CDEA00		CALL	TAPODN
2350i	F153 E1		POP	HL,ABORT
2360i	F154 0A60B		LD	A,HL
2370i	F156 7E		LD	A,HL
2380i	F157 E5	NSV:	PUSH	HL
2390i	F158 CDE000		CALL	TAPOUT
2400i	F158 E1		POP	HL
2410i	F15C 3841		JP	C,ABORT
2420i	F15E 23		INC	HL
2430i	F15E E5		INC	HL
2440i	F160 AF		XOR	A
2450i	F161 1160F0		LD	DE,RTOP
2460i	F164 ED52		SBC	HL,DE
2470i	F166 7C		LD	A,H
2480i	F167 B5		OR	L

リスト4b(その4)

2500i	F16B E1		POP	HL
2510i	F169 20EB		CALL	NZ,NSV
2520i	F16B CDF000		CALL	TAPODF
2530i	F16E C3A0F0		RTOP	
2540i	F171 1198F2	LOAD:	LD	DE,SVMSG
2550i	F174 CD9A60		CALL	CHSET
2560i	F177 CD9F00		CALL	HL,(BOTTOM)
2570i	F17A 2A4BFC		PUSH	HL
2580i	F17D E5		CALL	TAPOIN
2590i	F17E CDE100		CALL	HL,ABORT
2600i	F182 3B1B		JP	HL,ABORT
2610i	F184 E5	NLD:	PUSH	HL
2620i	F185 CDE400		CALL	TAPOIN
2630i	F18B E1		CALL	HL
2640i	F189 3B14		JP	C,ABORT
2650i	F18B 77		LD	(HL),A
2660i	F18C 23		LD	HL
2670i	F18E AF		PUSH	HL
2680i	F18E AF		XOR	A
2690i	F18F 1160F0		LD	DE,RTOP
2700i	F192 ED52		SBC	HL,DE
2710i	F194 7C		LD	A,H
2720i	F194 7C		LD	A,H
2730i	F194 7C		LD	A,H
2740i	F194 E1		LD	HL
2750i	F194 E1		LD	HL
2760i	F197 20EB		JP	NZ,NLD
2770i	F199 CDE700		CALL	TAPODF
2780i	F19C C3A0F0		JP	RTOP
2790i	F19F 11E4F2	ABORT:	LD	DE,SVMSG
2800i	F1A2 CD9A60		CALL	DE,RTOP
2810i	F1A2 CD9A60		CALL	CHSET
2820i	F1A8 C3A0F0		CALL	RTOP
2830i	F1AB F5	WAIT:	PUSH	AF
2840i	F1AC 3A79F3		LD	A,(WAITF)
2850i	F1B0 20FD	LLL:	DEC	A
2860i	F1B2 C9		POP	A,LLL
2870i	F1B3 C9		RET	AF
2880i	F1B4 1114F3	SPEED:	LD	DE,SVMSG
2890i	F1B7 CD9A60		CALL	CHSET
2900i	F1B8 CDE700		CALL	HL
2910i	F1B8 CDE700		CALL	HL
2920i	F1B8 CDE700		CALL	HL
2930i	F1B8 CDE700		CALL	HL
2940i	F1B8 CDE700		CALL	HL
2950i	F1B8 CDE700		CALL	HL
2960i	F1B8 CDE700		CALL	HL
2970i	F1B8 CDE700		CALL	HL
2980i	F1B8 CDE700		CALL	HL
2990i	F1B8 CDE700		CALL	HL


[illegible]



迷路メイズ

最近、迷路ゲームをあちこちで見かけますが、皆さんは、迷路ゲームは好きですか？ なかなかユニークな迷路がたくさんあっておもしろいのですが、いざ自分で迷路を作ろうとすると、これがなかなかたいへんですね。できた！ と思ってやってみると迷路にならず、いきづまりなんてこともあれば、なあ～んだ、こんなに簡単なの、なんてこともありますよね。

そこで、これを MSX にやらせちゃおうというのが、このゲームです。どうせなら、画面いっぱいにつけて、より複雑にした方がおもしろいですね。では、始めましょう。君は、何秒でぬけることができるでしょうか？



遊び方

STICK (0)
STICK (1)

まず、リスト 5 を入力して RUN してください。すると迷路を作り始めます。この迷路作成にはしばらく時間がかかりますので、リストでもながめて一息入れてください。

迷路が完成しますと、ピッと音がしてゲームスタートとなります。左上の赤色の四角があなたで、出口は右下となっています。このゲームでは、本体のカーソルキーとジョイスティックの No.1 を同時に使えますので、どちらでもゲームの途中で変えることができます。うちのグループでジョイスティックで遊んでいた人のところで本体のカーソルキーをいじったために、ひっぱたかれた人がおります。逆に 2 人で 1 人はジョイスティック、もう一人は本体のカーソルキーで動かすとおもしろいんじゃないでしょうか。


みごと迷路を脱出しますと、脱出するのにかった時間と、その時間に対するキツ〜イ評価とへんちくりんな音楽が流れます。このゲームのこつは、ゲームがスタートしても、すぐに動かさず、まず目で出口から逆にスタートまでをたどってからおもむろに動かすと、実にスムーズに抜けることができます。

プログラム

このプログラムの最大のポイントは、何と言っても迷路を作るアルゴリズムです。このアルゴリズムは、本書の「ハイパーチェイサー」のスネークパターン方式ではなく、俗称スキャン方式と呼ばれている方法を使用しています。まずこのアルゴリズムを解説しましょう。

さて、はじめに正方形を書いてその真中に点を打ちます。次にこの点を起点にして上下左右のどちらか一方に線を引きます (図 1)。これがこの方式の基本形です。次に図 2 のようにこの基本形を 4 つにしてみます。つまり 4 つの起点から同じように上下左右に線を 1 本だけ引くと、だんだん迷路らしくなってきました。しかし注意しないと、図 3 のように道がループを作ってしまうことがあります。そこで、次のような規則を作り、迷路を作成します。

1. 壁になる線を引くのは必ず一番上の段の点から始める



2. 一番上の段以外の点から壁になる線を引けるのは下右左の3つだけとする

3. 線はかさなってはいけない

つまり一番上の段の点をすべて処理してから下の段の処理を行うということ、一番上以外の段では上方向に線を引かないということで、これを守れば図3のようなループはできないことになります。

このアルゴリズムの特長は、考え方がちょっと難しいのですが、高速であるということと道のうねりが多いので、直感的に解くことができないということです。実際のプログラムでは270～570行が迷路作成の部分ですので、読んでみてください。ちょっと変更すれば100×100の迷路もなんなく作ることができます。

DRAW

もう一つのこのプログラムの特長は、タイトルに **DRAW** 命令を使っていることで、130～190行で MAZE GAME というきれいなタイトルを作成しています。

POINT TIME

また、移動の際の壁の判断は、**POINT** 命令による色で判断を行っています。時間の表示は、**TIME** 変数の値を60で割ることにより近似的に秒数を計算しています。

このプログラムは16Kバイト以上のシステムで使用可能で、ディスクシステムでも OK です。


```
10 ' Super maze game
20 ' by Tamae tama
30 ' debug and advice by Haruka Takagi
40 ' copyright 1984 (C) NATS
50 ' copyright 1984 (C)
60 '     Seibundo-Shinkousha publishing
70 '
80 ' HENSU & SYSTEM INIT
90 '
100 SCREEN 2:COLOR 15,4,4
110 OPEN "GRP:" FOR OUTPUT AS #1
120 DEFINT A-Z:F=0:R=RND(-TIME)
130 DRAW"BM30,50L20D20R20
140 DRAW"BM40,70U20R20D10L20R10M60,70
150 DRAW"BM70,50R20L10D20
160 DRAW"BM120,70U20M+10,+10M+10,-10D20
170 DRAW"BM150,70U20R20D20U10L20
180 DRAW"BM180,50R20M-20,+20R20
190 DRAW"BM230,50L20D10R10L10D10R20
200 PRESET(50,100):PRINT#1,"By T.Tama & H.Takagi"
210 PRESET(50,130):PRINT#1,"Copyright 1984 NATS
220 PRESET(50,150):PRINT#1,"Copyright 1984 (C)
230 PRESET(50,160):PRINT#1,"Seibundo-Shinkousha
240 PRESET(34,180):PRINT#1,"PUSH TRIGER or SPC KEY !!
250 IF STRIG(0)=-1 OR STRIG(1)=-1 THEN 260 ELSE 250
260 SCREEN 3
```

```

270 '
280 ' "WAKU" WRITE
290 '
300 LINE(8,0)-(240,184),15,B
310 PSET(12,4),8
320 PSET(236,184),0
330 '
340 ' "UE EDA" WRITE
350 '
360 FOR X=16 TO 236 STEP 8
370 PSET(X,8),15
380 ON 1+4*RND(1) GOTO 390,410,420,430
390 IF F THEN ON 1+3*RND(1) GOTO 410,420,430
400 PSET(X-4,8),15:GOTO440
410 PSET(4+X,8),15:F=1:GOTO440
420 PSET(X,12),15:F=0:GOTO440
430 PSET(X,4),15:F=0
440 NEXT
450 '
460 ' "NAKA EDA" WRITE
470 '
480 FOR Y=16 TO 180 STEP 8
490 F=0
500 FOR X=16 TO 236 STEP 8
510 PSET(X,Y),15
520 ON 1+3*RND(1) GOTO 530,550,560
530 IF F THEN ON 1+2*RND(1) GOTO 550,560
540 PSET(X-4,Y),15:GOTO 570
550 PSET(4+X,Y),15:F=1:GOTO 570
560 PSET(X,4+Y),15:F=0
570 NEXT:NEXT

```


```
580 '
590 ' GAME START
600 '
610 X=12:Y=4
620 BEEP:BEEP
630 TIME=0
640 '
650 ' KEY IN
660 '
670 FOR I=0 TO 1
680 J=STICK(I)
690 IF 0=J THEN NEXT:GOTO 670
700 '
710 ' MOVE
720 '
730 IF 3=J AND 15<>POINT(X+4,Y) THEN XX=4:YY=0:GOTO 770
740 IF 7=J AND 15<>POINT(X-4,Y) THEN XX=-4:YY=0:GOTO 77
0
750 IF 5=J AND 15<>POINT(X,Y+4) THEN XX=0:YY=4:GOTO 770
760 IF 1=J AND 15<>POINT(X,Y-4) THEN XX=0:YY=-4 ELSE 67
0
770 PRESET(X,Y):X=X+XX:Y=Y+YY
780 PSET(X,Y),8
790 IF X=236 AND Y=184 THEN BEEP:BEEP ELSE FOR I=0 TO 3
00:NEXT:GOTO670
```



```

800 '
810 ' GAME END
820 '
830 T=TIME:FOR I=0 TO 2000:NEXT
840 SCREEN 1:KEY OFF
850 LOCATE 9,8:PRINT"GAME OVER"
860 LOCATE 4,10:PRINT"しょう し`かん";T¥60;"ヒ`ョウ"
870 PLAY"05M1112S10T240EDCDE","04M1112S10T240ACEACE"
880 PLAY"05M1112S10T240EDCDE","04M1112S10T240ACEACE"
890 PLAY"05M1112S10T240EDCEDCCDEFO6C","04M1112S10T240AB
    CABCEFG"
900 IF PLAY(0) THEN 900
910 QA=T¥60
920 IF QA<=30 THEN PRINT "      すこ`い にんけ`んし`ゃ ない!!!":GOTO
    980
930 IF QA<=45 THEN PRINT "      たいした もので`す よ ホント!!!":GOTO
    980
940 IF QA<=60 THEN PRINT "      まあ まあ て`すね たた`のひと  ":GOTO
    980
950 IF QA<=90 THEN PRINT "      あなた,にんけ`ん し`ゃないよ!!!":GOTO
    980
960 IF QA<=135 THEN PRINT "      ノロマ!! ト`シ`!! カメ!!  ":GOTO
    980
970 PRINT "      もう,にんけ`ん やめたほうか` いい"
980 LOCATE 0,15:PRINT "      Try again? (y/n)"
990 A$=INKEY$
1000 IF A$="Y" OR A$="y" THEN RUN
1010 IF A$="N" OR A$="n" THEN END ELSE GOTO 990

```




ハイパーチェイサー



毎日、夜遅くまでマイコンの前に座り込んでいる僕は何をやっているかって、なんてことはない。いっしょうけんめいゲームを作っているんだ。おもしろいゲームを目ざし、どこかのプロコン（プログラムコンテスト）に入賞する夢を見ている。

時は真夜中12時。どこからか時計のボーン、ボーンという音が聞こえてくる。やっと完成間近のプログラムを、テストランさせようと、ファンクションキーを押した。と、その瞬間、CRT画面がスパークし、気がつくとまったく見知らぬところにいる。あれ〜！僕はたしかプログラムを作っていて、テストランを始めたはずだが…。

な〜んで話が始まるんだけど、ようするに君は、プログラムの迷路に入り込んだ。よくみると、出口はある。しかし、何かこちらに向かってくるぞー。あ〜！バグ虫だ。そこで君に指示する。このバグ虫に捕まらないように、この分岐あり、いきづまりありのプログラム迷路から脱出して欲しい。バグ虫は壁をつぎつぎに破壊してくる。さて君は何秒で脱出できるか。



遊び方

**RUN
INKEY \$
RND (1)**

このゲームは画面いっぱいに作られた迷路を、バグ虫につかまらないように右下の出口から脱出するものです。


内容はいたって簡単なのですが、バグ虫の追跡をうまくふり切らないと簡単につかまってしまうぞ〜。コツとしては、バグ虫をうまく引き付け、逆にバグ虫の破壊した壁の道を通して脱出することなんだ。とはいっても簡単に脱出するのはたいへん難しいんですが。

まず、リスト6を入力し、RUNするとキー入力待ちとなりますので、何でも良いからキーを押してください。この入力待ちになるのは、BASICの乱数の初期値をよくかきまぜるためのもので、これがないと電源入力後は毎回同じ迷路が作られてしまいます。

次に迷路を作りはじめますが、これに少々時間がかかりますのでじっと画面を見て脱出路を考えるなり、お茶でもいっぱい飲むなりして待っていてください。迷路が書き終わると音楽が鳴り、ゲームスタートとなります。

左上で点滅しているのがあなたで、バグ虫は右下（出口）から出てきます。移動は本体のカーソルキーでもジョイスティック（ただしNo.1に接続してください）のどちらでもけっこうです。

バグ虫からのがれ、うまく脱出できれば、所用時間が表示されます。逆にバグ虫につかまると、その時点でゲームオーバーになってしまうので、慎重に!!




プログラム

SCREEN 3

このプログラムは、他のゲームと異なりマルチカラーモード (SCREEN 3) を使用しています。このモードは、64×48ドットで低解像度のグラフィックとも言え、ドットが大きいのが特長ですが、逆にドットごとに色を変えることが可能です。迷路を作るには最適なわけです。

さて、このプログラムの中心は何と言っても、迷路の作成フローです。迷路の作り方は何種類もあり、この方法は、迷路を作っていく様がヘビに似ているのでスネークパターンなどとも呼ばれています。つまり迷路の左上の始点から乱数で方向を決定しその方向に壁または、他の道につきあたり、動けなくなるまで道をのぼしていく方法です。

この方法を図式化すると図1のようになります。この図からわかるように、この方式は迷路が枝状に分かれるため複雑なものとなり、かつスタートから出口までの道が1つしかないという点に特長があります。このアルゴリズムは、方向決定および作成を320~440行で、また壁の判断を480行で行っています。迷路の作り方は本書のスーパーメイズで違った方法を使用していますので、そちらもあわせて見てください。



以上で迷路が完成し、次はゲームスタートとなるわけです。ゲーム中のポイントは「追いかけ」のプログラムで、ここでは人間とバグ虫の座標を最短とするようにバグ虫を動かしています。つまり、人間の座標を (X, Y)、バグ虫の座標を (A, B) とすると、

$$|(X-A)| + |(Y-B)|$$

すなわち、 $ABS(X-A) + ABS(Y-B)$ が最小となるように、バグ虫の X 軸、Y 軸方向の座標を決めているわけです。これはプログラムでは 1070~1130 で行っています。

この部分はフィードバックを利用しているので、プログラムを追って行くとちょっとわかりにくい点があるかもしれません。また、バグ虫は壁があるとその壁をぶち破って通って行くわけですが、そのときに音を出すようにしています。壁かどうかの判断は 1140 行を見てもらえばわかるように、**POINT** 命令を使って、その色が壁の色かどうかで壁の判断をしています。

さて、**SCREEN 3** モードでは、画面にテキストを表示するとおもしろいことになります。たとえば、

```
10 SCREEN 3 : CLS : OPEN "GRP : " FOR OUTPUT
   AS # 1
20 PRINT # 1, "MAZE"
30 GOTO 30
```

というプログラムを実行してみてください。画面には、えらく大きい文字が表示されたでしょう？ 実は、バグ虫に捕まったときに表示される **YOU KILLED!!** という大きな文字は、これを利用してあるわけです。逆にこのモードでは、テキストモード時のような小さな文字は出せませんので、スコア表示などは、ちょっとムリのようなのです。

なお、このプログラムは 16K バイト以上のシステムで使用可能でディスクが付いていても OK です。

POINT

```

10 ' Highper chaser
20 ' by Mika Tada
30 ' debug and advice by Haruka Takagi
40 ' copyright 1984 (C) NATS
50 ' copyright 1984 (C) Seibundo-Shinkousha publishing
60 DEFINT A-Z:OPEN "GRP:" FOR OUTPUT AS #1
70 DIM M(31,23):KEY OFF
80 SCREEN 1:COLOR 15,0,0:CLS
90 PRINT "*****";
100 PRINT " *           Highper chaser           *";
110 PRINT " *                                           *";
120 PRINT " *       by M.Tada & H.Takagi       *";
130 PRINT " *       copyright 1984 (C) NATS*";
140 PRINT " *       copyright 1984 (C)         *";
150 PRINT " *       Seibundo-Shinkousha         *";
160 PRINT "*****";
170 PRINT :PRINT
180 PRINT "           Hit any key to start"
190 IF INKEY$="" THEN A=RND(1):GOTO 190
200 SCREEN 3,0,0:COLOR 15,0,0
210 LINE (0,0)-(255,191),3,BF
220 FOR X=0 TO 31:M(X,0)=1:M(X,23)=1:NEXT
230 FOR Y=0 TO 23:M(0,Y)=1:M(31,Y)=1:NEXT
240 PSET (8,4),1:PSET (244,176),1:MX=240:MY=176
250 M(1,1)=1:M(30,22)=0
260 FOR LX=1 TO 30:FOR LY=1 TO 22
270 X=LX*8:Y=LY*8:X1=LX:Y1=LY
280 IF M(X1,Y1-1)+M(X1,Y1+1)+M(X1-1,Y1)+M(X1+1,Y1)=4 TH
    EN NEXT :NEXT :GOTO 490
290 IF M(X1,Y1)=0 THEN NEXT :NEXT :GOTO 490
300 GOSUB 320
310 NEXT :NEXT :GOTO 490

```

```


320 '
330 '      MEIROKAKU SUBROUTINE !
340 '
350 ON 1+4*RND(1) GOTO 360,380,400,420
360 IF M(X1+1,Y1) THEN 320
370 LINE (X,Y)-(X+8,Y),1,BF:M(X1+1,Y1)=1:X=X+8:GOTO 440
380 IF M(X1-1,Y1) THEN 320
390 LINE (X,Y)-(X-8,Y),1,BF:M(X1-1,Y1)=1:X=X-8:GOTO 440
400 IF M(X1,Y1+1) THEN 320
410 LINE (X,Y)-(X,Y+8),1,BF:M(X1,Y1+1)=1:Y=Y+8:GOTO 440
420 IF M(X1,Y1-1) THEN 320
430 LINE (X,Y)-(X,Y-8),1,BF:M(X1,Y1-1)=1:Y=Y-8
440 X1=X/8:Y1=Y/8:' POINTER SET
450 '
460 '      MAWARI WA KABE KA ?
470 '
480 IF M(X1,Y1-1)+M(X1,Y1+1)+M(X1-1,Y1)+M(X1+1,Y1)=4 TH
    EN RETURN ELSE 320

```

```

490 '
500 '   MEIRO DASHUTSU !
510 '
520 D1$="BAGFEDCO6BAGFEDCO5GAGFEDC"
530 D2$="RBAGFEDCO6BAGFEDCO5GAGFED"
540 D3$="RRBAGFEDCO6BAGFEDCO5GAGFE"
550 Z$="O7L32T200V14"
560 PLAY Z$,Z$,Z$
570 PLAY D1$,D2$,D3$
580 X=8:Y=4:DM=0:PSET (8,4),4
590 LP=0:KL=0
600 IF STICK(0)=5 OR STICK(1)=5 THEN TIME=0:GOTO 610 EL
    SE PSET (8,4),DM:DM=DM XOR 4:GOTO 600
610 PRESET (X,Y):Y=Y+4:PSET (X,Y),4 :LOOP=0
620 IF STICK(0)=1 OR STICK(1)=1 THEN 690
630 IF STICK(0)=5 OR STICK(1)=5 THEN 750
640 IF STICK(0)=3 OR STICK(1)=3 THEN 800
650 IF STICK(0)=7 OR STICK(1)=7 THEN 850
660 LP=LP+1:IF LP=3 THEN LP=0
670 IF LP<>0 THEN 620
680 GOSUB 1060:IF KL=1 THEN GOTO 1040 ELSE 620

```



—リスト6(その4) ハイパーチェイサー—

```
690 '
700 '   UP KEY ON !
710 '
720 IF Y<8 THEN 620
730 IF POINT(X,Y-4)=3 THEN 620
740 PRESET (X,Y):Y=Y-4:GOTO 900
750 '
760 '   DOWN KEY ON !
770 '
780 IF POINT(X,Y+4)=3 THEN 620
790 PRESET (X,Y):Y=Y+4:GOTO 900
800 '
810 '   RIGHT KEY ON !
820 '
830 IF POINT(X+4,Y)=3 THEN 620
840 PRESET (X,Y):X=X+4:GOTO 900
850 '
860 '   LEFT KEY ON !
870 '
880 IF POINT(X-4,Y)=3 THEN 620
890 PRESET (X,Y):X=X-4
```

```

900 PSET (X,Y),4
910 IF KL=1 THEN 1040
920 IF X>243 AND Y>175 THEN 930 ELSE 660
930 BEEP:BEEP:BEEP
940 TM=TIME:FOR X=0 TO 31:FOR Y=0 TO 23:M(X,Y)=0:NEXT :
    NEXT
950 SCREEN 0:CLS
960 FOR DM=0 TO 30
970 LOCATE 0,10:PRINT "          Congratulation!!!!"
980 LOCATE 0,10:PRINT "          "
990 NEXT
1000 PRINT USING "####.## SEC カカリマシタ !! ";TM/60:PLAY "C
    ", "G", "E":PLAY "C16D16E16":PLAY "C", "G", "E"
1010 PRINT :PRINT "TRY AGAIN ( Y OR N ) "
1020 A$=INKEY$
1030 IF A$="Y" OR A$="y" THEN ERASE M:DIM M(31,23):GOTO
    80 ELSE IF A$="N" OR A$="n" THEN 1050 ELSE 1020
1040 CLS:PSET (0,0):PRINT #1," YOU          KILLED          !!":P
    LAY "G16G+16G16.G+16.G32G+32. ":FOR TM=0 TO 5000:NEX
    T:SCREEN 0:GOTO 1010
1050 END

```





```

1060 IF MX=X AND MY=Y THEN KL=1:RETURN
1070 IF ABS(MX-X)<ABS(MY-Y)*1.18 THEN 1080ELSE 1090
1080 IF ABS(MY-Y)<>0 THEN IF MY>Y THEN 1100 ELSE 1110 E
    LSE 1090
1090 IF ABS(MX-X)<>0 THEN IF MX>X THEN 1120 ELSE 1130 E
    LSE 1080
1100 IF MY>8 THEN PRESET (MX,MY):MY=MY-4:GOTO 1140 ELSE
    1140
1110 IF MY<172 THEN PRESET (MX,MY):MY=MY+4:GOTO 1140 EL
    SE 1140
1120 IF MX>12 THEN PRESET (MX,MY):MX=MX-4:GOTO 1140 ELS
    E 1140
1130 IF MX< 240 THEN PRESET (MX,MY):MX=MX+4:GOTO 1140 E
    LSE 1140
1140 IF POINT (MX,MY)=0 OR POINT (MX,MY)=1 THEN GOTO 11
    50 ELSE PLAY"t255164ca"
1150 PSET (MX,MY),8
1160 IF PLAY(0) GOTO 1160
1170 IF X=MX AND Y=MY THEN KL=1 ELSE KL=0
1180 RETURN

```

カセットファイルズ プログラム



カセットテープというものは便利なもので、1本のカセットに何本もプログラムを入れておくことができますが、あんまり入れすぎたり、カセットの本数が増えてくると、どうしても中にいったい何が入っているのかわからなくなってしまう場合があります。また、ついうっかり内容を書き移しておくのを忘れてそれっきりになってしまったり……。

こんなときは、いちいちカセットからプログラムをロードして内容をチェックしていく方法がありますが、それは簡単には行きません。というのは、MSX では BASIC プログラム (**CSAVE** したもの)、機械語プログラム (**BSAVE** したもの) それに ASCII 形式のプログラムおよびデータ (**SAVE "CAS :"** など) と 3 種類の形式があり、それぞれ対応するロードのし方をしないと、エラーとなってしまいます。このプログラムは、このようなわずらわしい作業をしなくても、1本のカセットテープに入っているすべての内容 (ファイルと呼びます) を表示してくれるもので、表示される内容は、

1. ファイルの形式

BASIC か、機械語か ASCII 形式のプログラムまたはデータかを表示

2. ファイル名

カセットにセーブしたときのファイル名です


3. 機械語プログラムのパラメータ

機械語プログラムの場合は、さらにプログラムのスタート番地、終了番地、実行番地を表示します

というように、そのファイルの内容を知るのに必要な情報を表示してくれます。

もし、内容まで知りたい場合、この情報をもとに各プログラムに適したコマンドでロードし、その内容を見れば良いわけです。


またこの内容は、プリンターに出すこともできますので、プリンターを接続しておけばモニター・TVの前になくても、ほうっておけば1本のカセットの内容すべてをプリンターに出力してくれます。あとはプリンター用紙を見ながらカセットラベルに内容を書き込んで行けば良いわけです。



プログラムの原理

みなさん、一度はカセットにセーブされたプログラムの音を聴いたことがあると思いますが、まったくギャゴ・ギャゴとそうぞうしいね。ところで、この音を注意深く聴くと図1のようにになっていることに気付くと思います。この最初のギャというところは実はヘッ

図1



ドブロックと呼び、プログラムのファイル名が入っているのです。

この後のピーの後のギャグ・ギャグと長く続く音は、プログラムの本体で、特に機械語プログラムの場合は、このギャグ・ギャグの頭にプログラムのスタート番地、終了番地、実行番地が入っています。また、プログラムの形式の区別はヘッドブロックの先頭10バイトのデータから判断できます。

さて、カセットにどのような形でセーブされているのかはわかりましたので、後は各ファイルのヘッドブロックとメインブロックの先頭を読み込んで表示してやればプログラムが完成します。

しかし、カセットからのデータの読み込みだけは、とてもとてもBASICでは追い付くことができません。そこで、カセットからのデータの読み込みだけは機械語でやることにしました。

プログラム

プログラムは機械語プログラム(リスト7b)により、メモリーのE000番地以後に、各ファイルのヘッドブロックおよびメインブロックの先頭6バイト、計23バイト分をまず読み込み、次にBASICプログラムにより各表示を行っています。特に難しい命令は使用していませんので内容についてはリスト7aを参照してください。

なお機械語プログラムは、BASICプログラム中の670行以後にデータとして格納されており、BASICプログラムで、自動的にメモリーに作られるようになっています。よって、リスト7aのみを入力すればけっこうです。

プログラムの使い方

まずリスト7aを入力して、セーブしておきます。特に機械語プログラムを併用していますので、入力ミスがありますと暴走する可能性がありますから、必ずセーブしてください。

LPRINT

次にRUNしますと、タイトルが出て、プリンターに出力するかどうかを聞いてきますので、プリンターに出す場合はy、出さない場合はnを押します。そして、オートマチックモードにするかどうかを聞いてきます。

オートマチックモードとは、1つのファイルの表示を行った後、自動的に次のファイルを連続して調べて行くモードで、プリンターと併用することで、何もしなくても1本のカセット中のすべてのファイルを調べることができます。オートマチックモードではない場合は、1つのファイルの表示をするごとにストップし、次のファイルを調べるか、プログラムを終了するか聞いてきます。

オートマチックモードにする場合はy、そうでない場合はnを入力してください。これでプログラムが起動しますので、後はカセットテープを再生すれば、画面またはプリンターに各ファイルの情報が出力されます。このプログラムは、16Kバイトシステム以上で使用可能で、ディスクがある場合は、ディスクをはずしてください。

使用例

```
file is BASIC program
filename=Cafp1

file is ASCII data or program
filename=Cafp2

filename=TO300:

filename=,1:PRI

filename=";
20

filename=JS=0:G

file is machine language
filename=CafpM
start address=C700
end address=F37F
entry address=C700
```


```
10 ' Cassette files program
20 ' by Haruka Takagi
30 ' copyright 1984 (C) NATS
40 CLEAR 100,&HDFFF
50 MAXFILES=0
60 SCREEN 0:CLS
70 PRINT "*****";
80 PRINT "*   Cassette files program Ver 1.0   *";
90 PRINT "*   by Haruka Takagi                   *";
100 PRINT "*   copyright 1984 (C) NATS           *";
110 PRINT "*   copyright 1984 (C)                 *";
120 PRINT "*   Seibundo-Shinkousha publishing  *";
130 PRINT "*****";
140 PRINT:PRINT:PRINT
150 PRINT "This program shows all of filename of your c
    assette tape"
160 PRINT
170 INPUT "list to printer (y/n)";DM$
180 IF DM$="y" THEN LP=1 ELSE LP=0
190 INPUT "automatically find next file (y/n)";DM$
200 IF DM$="n" THEN AF=0 ELSE AF=1
210 'main routine
220 CLS
230 AD=&HF000
240 READ D$:IF D$="end" THEN GOTO 270
250 D=VAL("&H"+D$):POKE AD,D
260 AD=AD+1:GOTO 240
270 DEF USR=&HF000
280 AD=&HE000
290 FOR I=0 TO &H16
300 POKE (AD+I),0
310 NEXT
320 PRINT "please tape start"
330 A=USR(0)
```

```

340 'error detect
350 D=PEEK(&HF024):IF D<>0 THEN PRINT "tape read error
    !! ":INPUT "retry (y) or end (n)";DM$:IF DM$<>"n" T
    HEN PRINT:GOTO 270 ELSE END
360 'build message data
370 AT=PEEK(&HE000)
380 AD=&HE00A
390 F$=""
400 D=PEEK(AD):AD=AD+1:IF AD=&HE011 THEN GOTO 420
410 F$=F$+CHR$(D):GOTO 400
420 ST=PEEK(AD)+PEEK(AD+1)*256
430 ST$=RIGHT$("000"+HEX$(ST),4)
440 AD=AD+2
450 EN=PEEK(AD)+PEEK(AD+1)*256
460 EN$=RIGHT$("000"+HEX$(EN),4)
470 AD=AD+2
480 SS=PEEK(AD)+PEEK(AD+1)*256
490 SS$=RIGHT$("000"+HEX$(SS),4)
500 'print message
510 PRINT
520 IF AT=&HEA THEN PRINT "file is ASCII data or progra
    m"
530 IF AT=&HD0 THEN PRINT "file is machine language"
540 IF AT=&HD3 THEN PRINT "file is BASIC program"
550 PRINT "filename=";F$
560 IF AT<>&HD0 THEN GOTO 610
570 PRINT "start address=";ST$
580 PRINT "end    address=";EN$
590 PRINT "entry address=";SS$

```

```
600 'print to printer
610 IF LP=0 THEN GOTO 700 ELSE LPRINT
620 IF AT=&HEA THEN LPRINT "file is ASCII data or progr
    am"
630 IF AT=&HD0 THEN LPRINT "file is machine language"
640 IF AT=&HD3 THEN LPRINT "file is BASIC program"
650 LPRINT "filename=";F$
660 IF AT<>&HD0 THEN GOTO 700
670 LPRINT "start address=";ST$
680 LPRINT "end   address=";EN$
690 LPRINT "entry address=";SS$
700 IF AF=1 THEN GOTO 270 ELSE PRINT :INPUT "next (n) o
    r end (e)";DM$
710 IF DM$<>"e" THEN GOTO 270 ELSE END
720 ' machine language data
730 DATA 3E,FF,32,24,F0,CD,E1,00
740 DATA 38,16,21,00,E0,06,17,E5
750 DATA C5,CD,E4,00,C1,E1,38,08
760 DATA 77,23,10,F3,AF,32,24,F0
770 DATA CD,E7,00,C9,end
```




```


1000:                ;Cassette files program
1010:                ;by Haruka Takagi
1020:                ;copyright 1984 NATS
1030:                ;
1040:    00E1 =        TAPION EQU    00E1H
1050:    00E4 =        TAPIN  EQU    00E4H
1060:    00E7 =        TAPIOF EQU    00E7H
1070:    E000 =        BUFF  EQU    0E000H
1080:                ;
1090:    F000           ORG    0F000H
1100:    F000 3EFF     START: LD     A,0FFH
1110:    F002 3224F0    LD     (EFLAG),A
1120:    F005 CDE100    CALL    TAPION
1130:    F008 3B16      JR      C,EXIT
1140:    F00A 2100E0    LD     HL,BUFF
1150:    F00D 0417      LD     B,17H
1160:    F00F E5        NEXT:  PUSH  HL
1170:    F010 C5        PUSH  BC
1180:    F011 CDE400    CALL    TAPIN
1190:    F014 C1        POP   BC
1200:    F015 E1        POP   HL
1210:    F016 3B08      JR      C,EXIT
1220:    F018 77        LD     (HL),A
1230:    F019 23        INC   HL
1240:    F01A 10F3      DJNZ   NEXT
1250:    F01C AF        XOR   A
1260:    F01D 3224F0    LD     (EFLAG),A
1270:    F020 CDE700    EXIT:  CALL  TAPIOF
1280:    F023 C9        RET
1290:    F024          EFLAG:  DEFS   1
1300:    F025          END

```

```

E000  BUFF      F024  EFLAG      F020  EXIT      F00F  NEXT
F000  START     00E4  TAPIN      00E7  TAPIOF    00E1  TAPION

```



西暦3001年、人類はある惑星からの攻撃を受けていた。そのころ地球は？
という、バルカン砲をつけただけという単純な宇宙船しかなかったの
だ!! その惑星の攻撃とは、UFO がエイリアンの卵を宇宙船の近くまで運
びふ化させるもので、ふ化したエイリアンが宇宙船にとり付くと、なんと
一瞬にして宇宙船を破壊してしまうというオソロシーものなのだ!!

さあ、君は地球を救うために宇宙船バランに乗り込むのだ、ナーンチャ
ッテ。



遊び方

PLAY STICK (0) STICK (1)

まず、リスト 8 を間違いのないようにシコシコと入力してください。次に RUN しますと、まさに宇宙サウンドと言った趣向でタイトルが表示されます。そして、ジョイスティックで遊ぶか、キーボードのカーソルキーで遊ぶか聞いてきますので、1 または 2 で答えてください。なお、ジョイスティックは No.1 側に接続してください。


次に画面がクリアされ、ゲームがスタートとなります。あとは、ひたすら十字型の照準を UFO に合わせて打ちまくるのみです。

さて、UFO は画面上から下にだんだん降りてくるわけですが、図のように下に降りるにつれ卵が出現し、さらに降りると卵がふ化し、エイリアンとなります。UFO が卵をだいている (?) 状態では、卵を打っても卵はすぐに再生されてしまいますので、UFO を攻撃するようにしてください。逆に UFO を破壊しますと、卵も壊れます。



また卵がふ化してエイリアンになっても、エイリアンはしばらくの間、UFO にくっついていきます。

この状態ではエイリアンは UFO のバリアに守られていて、エイリアンを破壊することはできませんので、同じように UFO を攻撃するようにしてください。さて、エイリアンが UFO から離れますと、うろうろとバランに取り付こうとしますので、必死になってエイリアンを攻撃してください。こいつに取り付かされると、一瞬にし

UFO の卵



エイリアン



て balan はクラッシュし、地球は占領されてしまいます。

ところで、UFO をどんどん攻撃して行きますと、最大 6 匹まで増えます。この状態で一定時間が経過しますと、ハデなサウンドと共に WARN!! (注意!!) という文字が表示されます。この状態になると、なんと UFO がどの高さにあってもすべてエイリアンに変化して balan を攻撃してきます。これを必死になって撃ち落とすと、再び UFO が一匹から始まります。

プログラム

プログラムは、けっこう大きなものですが、そのひとつひとつの作りはきわめて単純なものです。

まず 1500 行までが、いろんな変数、スプライトのデータ、効果音のデータなどの初期化を行う部分で、この行以後が、本体のプログラムとなっています。

本体のプログラムは、キー入力、UFO 移動、エイリアンの移動、UFO にタマが当たったときの処理、エイリアンにタマが当たったときの処理、WARN の処理、ゲームオーバーの処理を行っています。ほとんどの処理が、UFO、エイリアン、照準の座標移動とそれに属す処理ですが、WARN 処理だけはタイマー割り込みを使って行っています。このタイマー割り込みというのは、**ON INTERVAL GOTO** で行う処理のことで、この命令で指定した時間になると、BASIC が今どの行の処理を行っていても、指定した行に処理を移すことができます。これによって正確な時間に WARN 処理を行うことが可能となったわけです。

**ON~INTERVAL
GOTO**

SPRITE

さて、UFO やエイリアンそして照準ですが、これらは、すべてスプライトを使用しています。650 行からのデータを見てもらえばわかると思いますが、UFO でも 3 つのデータがあります。これは、遠近感を出すためのもので、UFO が上にある場合は小さく、逆に下になるほど大きい表示を行い、距離感を出しているわけです。

なお、このプログラムは 16K バイト以上のシステムで使用可能で、ディスクシステムでも OK です。

```

10 '*****
20 '**
30 '**          BARAN          **
40 '**
50 '** program: by Nami aoki    **
60 '** debug and advice:       **
70 '**          by Haruka Takagi **
80 '**
90 '** copyright 1984 (C) NATS   **
100 '** copyright 1984 (C)      **
110 '**      Seibundo-Shinkousha **
120 '**          publishing      **
130 '**
140 '*****
150 MAXFILES=1
160 DEFINT A-Z:A=RND(-TIME)
170 DIM MS(13),MA(13),V(10),U(10),UX(10),UY(10),VX(10),
    VY(10)
180 SC=0:HS=0
190 SCREEN 2,2,0:COLOR 15,0,0:CLS
200 OPEN "GRP:" FOR OUTPUT AS #1
210 D$="O5L32T200V8"
220 PLAY D$,D$,D$
230 D1$="O5CDEFGABO6CDEFGABO7CDEFGAB":D2$="O5RCDEFGABO6
    CDEFGABO7CDEFGA":D3$="O5RRDEFGABO6CDEFGABO7CDEFG"
240 PLAY D1$,D2$,D3$
250 PRESET (0,30),1

```

```

260 PRINT #1," *****"
270 PLAY D1$,D2$,D3$
280 PRINT #1," *           B A R A N           *"
290 PRINT #1," *                                           *"
300 PLAY D1$,D2$,D3$
310 PRINT #1," *   The Space fighter game   *"
320 PLAY D1$,D2$,D3$
330 PRINT #1," *       by N.Aoki & H.Takagi   *"
340 PRINT #1," *                                           *"
350 PLAY D1$,D2$,D3$
360 PRINT #1," *       Copyright 1984 NATS       *"
370 PLAY D1$,D2$,D3$
380 PRINT #1," *       Copyright 1984 (C)       *"
390 PLAY D1$,D2$,D3$
400 PRINT #1," *       Seibundo-Shinkousha       *"
410 PLAY D1$,D2$,D3$
420 PRINT #1," *****"
430 IF PLAY(0) GOTO 430
440 SOUND 7,&HC7:SOUND 8,&H10:SOUND 9,&H10:SOUND 10,&H1
    0:SOUND 13,1
450 PRINT #1,:PRINT #1,
460 ' SPRITE PATTERN SET
470 '
480 RESTORE
490 FOR I=1 TO 11
500 A$=""
510 FOR J=1 TO 32
520 READB$:A$=A$+CHR$(VAL("&H"+B$))
530 NEXT J:SPRITE$(I)=A$
540 NEXT I

```

リスト 8 (その 3) バラン

```
550 '
560 ' MUSIC DATA
570 '
580 FOR I=0 TO 13
590 READ A:MS(I)=A
600 NEXT
610 '
620 FOR I=0 TO 13
630 READ A:MA(I)=A
640 NEXT
650 '
660 ' DATA
670 '
680 '
690 ' DATA 1 UFO
700 DATA 0,0,0,0,0,1,9,F
710 DATA 9,0,0,0,0,0,0,0
720 DATA 0,0,0,0,0,80,90,F0
730 DATA 90,0,0,0,0,0,0,0
740 '
750 ' DATA 2 UFO
760 '
770 DATA 0,0,0,0,1,3,42,7F
780 DATA 43,5,0,0,0,0,0,0
790 DATA 0,0,0,0,80,C0,42,FE
800 DATA C2,A0,0,0,0,0,0,0
810 '
820 ' DATA 3 UFO
830 '
840 DATA 1,1,1,3,87,8F,FC,FF
850 DATA CF,87,9,10,20,0,0,0
860 DATA 80,80,80,C0,E1,F1,3F,FF
870 DATA F3,E1,90,8,4,0,0,0
```

```

880 '
890 ' DATA 4 INVADER
900 '
910 DATA 0, 3, 3, 7, 1F, 3F, 3F, 7F
920 DATA 7F, 71, FF, 11, 20, 40, 20, 10
930 DATA 0, C0, C0, E0, F8, FC, FC, FE
940 DATA FE, 8E, FF, 80, 4, 2, 4, 8
950 '
960 ' DATA 5 INVADER
970 '
980 DATA 0, 3, 3, 7, 1F, 3F, 3F, 7F
990 DATA 7B, 7D, FE, 89, 11, 21, 40, 80
1000 DATA 0, C0, C0, E0, F8, FC, FC, EE
1010 DATA DE, BE, FF, 91, 88, 84, 2, 1
1020 '
1030 ' DATA 6 SHOJUN
1040 '
1050 DATA 1, 1, 1, 1, 1, 1, 1, 1, FF
1060 DATA 1, 1, 1, 1, 1, 1, 1, 1, 1
1070 DATA 0, 0, 0, 0, 0, 0, 0, 0, FF
1080 DATA 0, 0, 0, 0, 0, 0, 0, 0, 0
1090 '
1100 '
1110 ' DATA 7 TAMAGO
1120 DATA 0, 0, 0, 7, F, 1F, 17, 17
1130 DATA 17, 1F, B, 7, 0, 0, 0, 0
1140 DATA 0, 0, 0, 80, C0, E0, E0, E0
1150 DATA E0, E0, CO, 80, 0, 0, 0, 0
1160 '
1170 ' DATA 8 BAKUHATSU
1180 '
1190 DATA 7B, 0, 0, 1F, 3F, 77, 3F, 17
1200 DATA ED, 3F, 5B, 1, 6F, 10, 18, A
1210 DATA 0, A0, EE, 94, 98, E8, 90, EE
1220 DATA F0, DE, B4, 80, B8, 98, 20, 30

```



—リスト 8 (その 5) バラン—

```
1230 '
1240 ' DATA 9 MISSILE L
1250 '
1260 DATA 0,0,0,0,0,0,1,3
1270 DATA 4,0,0,0,0,0,0,0
1280 DATA 0,0,0,0,0,00,00,80
1290 DATA 40,0,0,0,0,0,0,0
1300 '
1310 ' DATA 10 MISSILE M
1320 '
1330 DATA 0,0,0,0,0,0,01,03
1340 DATA 06,C,10,0,0,0,0,0
1350 DATA 0,0,0,0,00,00,00,80
1360 DATA C0,60,10,0,0,0,0,0
1370 '
1380 ' DATA 11 MISSILE S
1390 '
1400 DATA 0,0,0,0,00,00,01,03
1410 DATA 07,0F,1C,38,60,80,0,0
1420 DATA 00,00,00,00,00,00,00,80
1430 DATA C0,E0,70,38,C,2,0,0
1440 '
1450 ' MUSIC DATA 1
1460 '
1470 DATA 0,0,0,0,0,0,1,7
1480 DATA 16,16,16,100,100,0
1490 '
1500 '
1510 ' MUSIC DATA 2
1520 DATA 0,0,100,3,24,0,16,1
1530 DATA 0,10,10,90,2,13
```

```

1540 '
1550 ON INTERVAL =200 GOSUB 2890 2610 2910
1560 ON STRIG GOSUB 2610, 2610, 2610 2630
1570 PRINT #1, " JOY STICK (1) OR KEY BORD (2)"
1580 A$=INKEY$: IFA$="1" THEN JS=1 :GOTO 1600
1590 IF A$="2" THEN JS=0 ELSE GOTO 1580
1600 BEEP:BEEP:INTERVAL ON
1610 U(1)=1:UX(1)=RND(1)*32:UZ=1           :' UFO INI
    T.
1620 TI=0:UF=1:IV=1 :' TIMER INIT
1630 X=125:Y=80:CLS
1640 FOR I=0 TO 60:PSET (RND(1)*255,RND(1)*160),RND(1)*1
    4+2:NEXT
1650 FOR I=0 TO 80
1660 LINE (0+I,192)-(127,160),14
1670 LINE -(255-I,192),14
1680 NEXT
1690 FOR I=0 TO 40
1700 LINE (85+I,192)-(127,160),15
1710 LINE -(170-I,192),15
1720 NEXT:STRIG(JS) ON
1730 GOSUB 3240

```



リスト 8 (その 7) バラン

```

1740 '
1750 ' MAIN ROUTINE
1760 '
1770 ' KEY IN !
1780 '
1790 ST=STICK(JS)
1800 ON ST GOTO 1820,1830,1840,1850,1860,1870,1880,1890
1810 GOTO 1900
1820 Y=Y-8:      GOTO 1900
1830 X=X+8:Y=Y-8:GOTO 1900
1840 X=X+8:      GOTO 1900
1850 X=X+8:Y=Y+8:GOTO 1900
1860 Y=Y+8:      GOTO 1900
1870 X=X-8:Y=Y+8:GOTO 1900
1880 X=X-8:      GOTO 1900
1890 X=X-8:Y=Y-8:GOTO 1900
1900 IF X<0 THEN X=240
1910 IF X>240 THEN X=0
1920 IF Y<0 THEN Y=160
1930 IF Y>160 THEN Y=0
1940 PUT SPRITE 0,(X,Y),15,6
1950 '
1960 ' UFO GET
1970 '
1980 IF UF>0 THEN GOSUB 2080 2110
1990 '
2000 ' INVADER GET
2010 '
2020 IF IV>0 THEN GOSUB 2250 2280
2030 '
2040 ' WARN ROUTINE
2050 '
2060 IF WARN =1 THEN GOSUB 3180 3190
2070 GOTO 1740 1790

```

```

2080 '
2090 ' UFO ROUTINE
2100 '
2110 FOR I=1 TO UF
2120 UY(I)=UY(I)+RND(1)*U(I)*3*UZ
2130 UX(I)=UX(I)+TX+TX
2140 IF UY(I)>160 THEN UY(I)=150 : UZ=-1:BEEP
2150 IF UY(I)<0 THEN UY(I)=0: UZ=1
2160 IF UX(I)>240 THEN UX(I)=240:TX=-TX
2170 IF UX(I)<0 THEN UX(I)=0: TX=-TX
2180 IF U(I)=0 THEN 2230
2190 IF UY(I)<50 THEN U(I)=1
2200 IF UY(I)>50 THEN U(I)=2
2210 IF UY(I)>100 THEN U(I)=3
2220 PUT SPRITE 2+I, (UX(I), UY(I)), 14, U(I)
2230 NEXT
2240 RETURN
2250 '
2260 ' INVADER ROUTINE
2270 '
2280 FOR I=1 TO UF
2290 IF U(I)=1 THEN 2330
2300 IF U(I)=3 THEN V(I)=1
2310 VX(I)=UX(I):VY(I)=UY(I)
2320 IF U(I)=2 THEN PUT SPRITE 14+I, (VX(I), VY(I)+8), 10,
7:GOTO 2330
2330 NEXT
2340 FOR I=0 TO IV
2350 IF V(I)=0 THEN 2420
2360 VX(I)=VX(I)+TA
2370 IF VX(I)>240 THEN VX(I)=240:TA=-TA
2380 IF VX(I)<0 THEN VX(I)=0: TA=-TA
2390 VY(I)=VY(I)+RND(1)*8
2400 IF VY(I) > 160 THEN 2440 2470
2410 VZ=(VX(I)+VY(I)) MOD 2:PUT SPRITE 14+I, (VX(I), VY(I)
), 10, 4+VZ
2420 NEXT
2430 RETURN

```

```

2440 '
2450 ' game over
2460 '
2470 FOR T=0 TO 13:SOUND T,MS(T):NEXT
2480 FOR TM=0 TO 100:Z=RND(1)*15+1:COLOR ,Z,Z:NEXT
2490 COLOR ,0,0
2500 STRIG(JS) OFF :INTERVAL OFF:FOR I=0 TO10
2510 PRESET(0,80):PRINT #1,"          G A M E O V E R "
      :BEEP
2520 LINE (0,78)-(255,90),1,BF:BEEP
2530 NEXT
2540 PRESET (40,144):PRINT #1,"          TRY AGAIN (Y/N)"
2550 A$=INKEY$:IF A$="" THEN 2550
2560 IF A$="y" OR A$="Y" THEN SCREEN 2,2,0:SC=0:ERASE U
      X,UY,VX,VY,U,V:DIM UX(10),UY(10),VX(10),VY(10),U(10
      ),V(10):GOTO 460
2570 IF A$="n" OR A$="N" THEN END ELSE GOTO 2550
2580 GOTO 2550
2590 ' NICE ! hit ROUTINE
2600 '
2610 ' MISIILE TRIGGER ROUTINE
2620 '
2630 GOSUB 2990
2640 '
2650 ' UFO HIT !
2660 '
2670 FOR H=1 TO UF
2680 IF UX(H)< X+4+U(H) AND UX(H)>X-4-U(H) THEN AT=1 EL
      SE AT=0
2690 IF UY(H)< Y+4+U(H) AND UY(H)>Y-4-U(H) THEN AT=AT+1
2700 IF AT<2 THEN 2750
2710 PUT SPRITE 1,(X,Y),15,8
2720 FOR J=0 TO 13:SOUND J,MS(J):NEXT
2730 PUT SPRITE 1,(0,209):PUT SPRITE 2+H,(0,209):PUT SP
      RITE 14+H,(0,209)
2740 U(H)=1:UY(H)=0:UX(H)=RND(1)*255:SC=SC+10:GOSUB 324
      0
2750 NEXT H

```


```

2760 '
2770 ' INVADER HIT !
2780 '
2790 FOR H=1 TO IV
2800 IF VX(H)< X+8 AND VX(H)>X-8 THEN AT=1 ELSE AT=0
2810 IF VY(H)< Y+8 AND VY(H)>Y-8 THEN AT=AT+1
2820 IF AT<2 THEN 2870
2830 PUT SPRITE 1, (X,Y), 15, 8
2840 FOR J=0 TO 13: SOUND J, MS(J): NEXT
2850 PUT SPRITE 1, (0, 209): PUT SPRITE 14+H, (0, 209)
2860 VY(H)=0: VX(H)=RND(1)*255: SC=SC+50: V(H)=0: GOSUB 324
    0
2870 NEXT H
2880 RETURN
2890 ' TIMER INTERRUPT
2900 '
2910 TI=TI+1
2920 TX=RND(1)*10-5
2930 TA=RND(1)*10-5
2940 IF TI MOD 20=0 THEN UF=UF+1: IV=IV+1: U(UF)=1: IF UF>
    6 THEN 3070 3100
2950 RETURN
2960 '
2970 ' MISILE & MISIILE SOUND
2980 '
2990 SOUND 7, 254: SOUND 8, 15
3000 FOR IN=1 TO 255 STEP 3
3010 SOUND 0, IN: NEXT: SOUND 0, 0
3020 XA=(X-120)/9: YA=(Y-160)/9: XB=120: YB=160
3030 FOR A=1 TO 9: XB=XB+XA: YB=YB+YA
3040 PUT SPRITE 0, (XB, YB), 9, 9+YB/53
3050 NEXT
3060 RETURN


```

```

3070 '
3080 '  WARN !!
3090 '
3100 PRESET (100,8),8:PRINT#1,"WARN!!"
3110 FOR K=0 TO13:SOUND K,MA(K):NEXT
3120 UF=1:IV=6:FOR H=1 TO 6:V(H)=1:U(H)=0:PUT SPRITE 2+
    H, (0,209):VX(H)=RND(1)*240:NEXT
3130 SC=SC+1000:TI=0:U(1)=1:WARN=1
3140 'LINE (100,8)-(150,16),1,BF
3150 RETURN
3160 '
3170 '  WARN ROUTINE
3180 '
3190 FOR HJ=1 TO IV:IF V(HJ)=1 THEN 3210
3200 NEXT HJ: LINE (100,8)-(150,16),1,BF:IV=1:SC=SC+100
    0:WARN=0:GOTO 3230
3210 FOR K=0 TO 13:SOUND K,MA(K):NEXT
3220 IV=6
3230 RETURN
3240 IF SC>=HS THEN HS=SC
3250 SC$=RIGHT$("0000"+RIGHT$(STR$(SC),LEN(STR$(SC))-1)
    ,5):HS$=RIGHT$("0000"+RIGHT$(STR$(HS),LEN(STR$(HS))
    -1),5)
3260 LINE (0,0)-(255,8),0,BF:PRESET(0,0)
3270 PRINT #1,"  SCORE ";SC$;"      HI-SCORE ";HS$
3280 RETURN
    
```




15ゲーム



15ゲームという名前は知らなくても、ほとんどの人がやったことのあるゲームだと思います。そうです。15個の数字や絵の書いてあるコマを動かし、ばらばらだった数字や絵をきちんとならべなおすゲームです。

コンピューターでやる場合は、数字ではつまりませんし、グラフィックだと、目がちらついてきて長時間遊ぶのはきついで、色をぬったタイルを使うことにしました。


また、最初の「かきまぜ」や完成までに何回コマを動かしたかなどは、すべてコンピューターがやってくれますので、人間様はゲームのみに集中することができます。



遊び方

まずリスト9を入力してください。次に RUN しますと、画面がクリアされてタイトルが出てジョイスティックを使うかどうか聞いてきますので、ジョイスティックを使う場合は y、本体のカーソルキーで遊ぶ場合は n を入力してください。

次に **How many times shuffling ?** (何回かきまぜますか) と聞いてきますので、かきまぜる回数を入力してください。だいたい初級で50回、中級で100回、上級で200回といったところでしょうか。あまりかきまぜると、にっちもさっちもいなくなってしまうます。かきまぜる回数を入力すると、まず完成したところの絵が出て、かきまぜを始め、かきまぜが完了するとゲームがスタートします。後は、指でコマをかちゃかちゃと動かすのと同じ要領でカーソルキーでコマを動かす方向を指示してやればコマが1つずつ動きます。



プログラム

プログラムは大きく4つの部分から作られています。

まず、コマをかきまぜる部分 (460行からの MIX UP) は、メモリー上にあるゲーム盤のコマを How many times suffling ? で答えた回数かきまぜる部分で、かきまぜのキーワードは520行の **RND** 関数による乱数によって行っています。

次にコマを動かす部分 (690行からの KOMA MOVE) ですが、ここでは、**STICK** 関数を使ってキー入力またはジョイスティックを読

RND

STICK

SWAP

み指定されたコマとコマのない部分を **SWAP** 命令を使って交換し、表示を行います。これはたとえば上から下にコマを動かしたい場合、上のコマとコマのない部分の場所を交換すると、コマが上から下に動いたように見えることを利用しているわけです。

ON GOTO

3 番目はコマの表示です。この部分はマイコン内にあるコマの番号に対応した絵を書くもので、コマの番号は完成時左上から右へ 0、1、2……となっていて、右下のコマのない部分が15番目になっています。ここでは、**ON GOTO** 命令を使って絵を描くルーチンに分岐するようになっていきますので、それぞれの番号に対応した飛び先に、自分の好きな絵を描くようにプログラムを変更すれば、あなただけの15ゲームができます（1090～1290行）。

最後は、絵が完成したか？ というのを判定する部分です。ここは先ほど説明したコマの番号を見て判定しています。つまり、コマが 0、1、2……14、15と並んでいれば、完成というわけですね。これは、820行からの GAME COMPLETE ?で行っています。

TIME

さてこのゲームは、時間よりもコマの移動回数の少なさを競うようになっていますが、これだけではつまらない、という人は時間もカウントし、両方の値によって評価を下すように変更するとおもしろいかもしれません。もちろん、時間は **TIME** 変数を使うわけです。

なお、このプログラムは、16Kバイト以上のシステムで使用可能で、ディスクシステムでも OK です。

```

10 ' 15 game
20 ' by Tamae Tama
30 ' debug and advice by Haruka Takagi
40 ' copyright 1984 (C) NATS
50 ' copuright 1984 (C)
60 '      Seibundo-Shinkousha publishing
70 '
80 ' SYSTEM & HENSU INIT
90 '
100 DEFINIT A-Z:KEY OFF
110 T$="15 GAME  By TAMAE"
120 C$="Copyright 1984 NATS"
130 SCREEN 1,0,0:CLS
140 PRINT "*****";
150 PRINT "          15 Game          *";
160 PRINT "          *";
170 PRINT "      by T.Tama & H.Takagi      *";
180 PRINT "      copyright 1984 (C) NATS    *";
190 PRINT "          copyright 1984 (C)    *";
200 PRINT "          Seibundo-Shinkousha    *";
210 PRINT "*****";
220 LOCATE 5,15:INPUT "joystick(y/n)";A$
230 IF A$="Y" OR A$="y" THEN J=1
240 LOCATE 1,17:INPUT "How many times shuffling";SU
250 SCREEN 2:COLOR 15,4,4
260 OPEN "GRP:" FOR OUTPUT AS #1
270 '
280 ' KOMA INIT
290 '
300 CLS:FOR Y=0 TO 3
310 FOR X=0 TO 3
320 K(X,Y)=X+Y*4
330 NEXT
340 NEXT
350 PRESET(48,12):PRINT#1,T$
360 PRESET(40,20):PRINT#1,C$
370 BX=3:BY=3

```

```
380 '
390 ' DRAW COMPLETE PICTUR
400 '
410 FOR Y=0 TO 3
420 FOR X=0 TO 3
430 GOSUB 1060
440 NEXT
450 NEXT
460 '
470 ' MIX UP
480 '
490 PSET (50, 160), 0: RN=RND (-TIME)
500 PRINT #1, "タタ`イマ カキマセ`テイマス"
510 FOR I=0 TO SU
520 RN=RND (1)*4
530 IF RN=0 AND BY<>0 THEN SWAP K (BX, BY), K (BX, BY-1): BY=
    BY-1: GOTO 570
540 IF RN=1 AND BY<>3 THEN SWAP K (BX, BY), K (BX, BY+1): BY=
    BY+1: GOTO 570
550 IF RN=2 AND BX<>0 THEN SWAP K (BX, BY), K (BX-1, BY): BX=
    BX-1: GOTO 570
560 IF RN=3 AND BX<>3 THEN SWAP K (BX, BY), K (BX+1, BY): BX=
    BX+1 ELSE 520
570 NEXT
580 CLS
590 FOR Y=0 TO 3
600 FOR X=0 TO 3
610 GOSUB 1060
620 NEXT
630 NEXT
640 MOVE=1
650 PRESET (48, 12): PRINT #1, T$
660 PRESET (40, 20): PRINT #1, C$
670 PRESET (50, 160)
680 PRINT #1, "1 かいめ"
```

```


690 '
700 ' KOMA MOVE
710 '
720 I=STICK(J)
730 IF I=5 AND BY<>0 THEN BEEP:SWAP K(BX,BY),K(BX,BY-1)
      :BY=BY-1:X=BX:Y=BY:GOSUB1060:Y=Y+1:GOSUB 1060:GOTO
      780
740 IF I=1 AND BY<>3 THEN BEEP:SWAP K(BX,BY),K(BX,BY+1)
      :BY=BY+1:X=BX:Y=BY:GOSUB1060:Y=Y-1:GOSUB 1060:GOTO
      780
750 IF I=3 AND BX<>0 THEN BEEP:SWAP K(BX,BY),K(BX-1,BY)
      :BX=BX-1:X=BX:Y=BY:GOSUB1060:X=X+1:GOSUB 1060:GOTO
      780
760 IF I=7 AND BX<>3 THEN BEEP:SWAP K(BX,BY),K(BX+1,BY)
      :BX=BX+1:X=BX:Y=BY:GOSUB1060:X=X-1:GOSUB 1060:GOTO
      780
770 GOTO 720
780 MOVE=1+MOVE
790 LINE (152,168)-(52,160),0,BF
800 PRINT #1,MOVE;"かいめ";

```

```

810 '
820 ' GAME COMPLETE ?
830 '
840 FOR Y=0 TO 3
850 FOR X=0 TO 3
860 IF 4*Y+X<>K(X,Y) THEN 720
870 NEXT
880 NEXT
890 PRINT #1," Nice !!"
900 PRESET(16,170)
910 Z$="O5L32T200V15R8"
920 D1$="CDEFGABAGFEDCABCD"
930 D2$="R4CDEFGABAGFEDCABC"
940 D3$="R4R4CDEFGABAGFEDCAB"
950 PLAY Z$,Z$,Z$
960 FOR TM=0 TO 1:PLAY D1$,D2$,D3$:NEXT
970 IF PLAY(0) THEN 970
980 PRINT #1,"EXIT TO GAME CTRL+STOP KEY
990 PRESET(16,178)
1000 PRINT #1,"REPLAY : TRIGER or SPC KEY
1010 IF STRIG(J)=0 THEN 1010 ELSE 300
1020 GOTO 1020

```



```

1030 '
1040 ' DRAW KOMA
1050 '
1060 XL=48+32*X:YL=30+32*Y
1070 LINE (XL,YL)-(32+XL,32+YL),15,BF
1080 LINE (XL,YL)-(32+XL,32+YL),1,B
1090 ON K(X,Y)+1 GOTO 1100,1110,1120,1130,1140,1150,116
    0,1170,1180,1190,1200,1210,1220,1230,1240,1250
1100 C=3:GOTO 1290
1110 C=3:GOSUB 1270:C=4:GOTO 1290
1120 C=4:GOSUB 1270:C=5:GOTO 1290
1130 C=5:GOTO 1270
1140 C=3:GOSUB 1280:C=6:GOTO 1290
1150 C=3:GOSUB 1260:C=6:GOSUB1270:C=4:GOSUB 1280:C=7:GO
    TO 1290
1160 C=4:GOSUB 1260:C=7:GOSUB1270:C=5:GOSUB 1280:C=8:GO
    TO 1290
1170 C=5:GOSUB 1260:C=8:GOTO 1270
1180 C=6:GOSUB 1280:C=9:GOTO 1290
1190 C=6:GOSUB 1260:C=9:GOSUB1270:C=7:GOSUB 1280:C=10:G
    OTO 1290
1200 C=7:GOSUB 1260:C=10:GOSUB1270:C=8:GOSUB 1280:C=12:
    GOTO 1290
1210 C=8:GOSUB 1260:C=12:GOTO1270
1220 C=9:GOTO 1280
1230 C=9:GOSUB 1260:C=10:GOTO 1280
1240 C=10:GOSUB 1260:C=12:GOTO 1280
1250 LINE (XL,YL)-(32+XL,32+YL),1,BF:RETURN
1260 LINE (XL,YL)-(10+XL,10+YL),C,BF:RETURN
1270 LINE (XL,22+YL)-(10+XL,32+YL),C,BF:RETURN
1280 LINE (22+XL,YL)-(32+XL,10+YL),C,BF:RETURN
1290 LINE (22+XL,22+YL)-(32+XL,32+YL),C,BF:RETURN


```

もぐらたたき



昔はゲームセンターでよくもぐらたたきをハンマーでひっぱたいっている人を見かけましたが、最近はこの手のゲームはあまり見ませんね。

このゲームは、ゲームセンターのモグラたたきをシミュレートするものですが、あくまでも画面に出てくるモグラを、キーを押すことで「たたく」わけですから、MSXをとんかちでたたくようなことはやめておいたほうが良いと思います……やっぱり。




遊び方

まずリスト10を入力して、RUN します。タイトル画面が出て、難易度を聞いてきます。1を入力するとモグラの動きが一番速く、3が一番遅くなります。また3でもまだ速い！ という人は3以上を入れても受け付けるように作ってあります。とは言っても、しょせん BASIC です、それほど速くはないはずです。難易度を入力すると次はゲームがスタートします。

画面には5つのモグラの穴が表示され、そのそばにF1、F2などと書かれておりますが、これがファンクションキーの番号に対応しており、対応するファンクションキーを押せばその穴をたたくようになっています。

モグラは、ちょっと顔を出すときと、完全に出したとき、そして消えかけてる時の3通りがあり、ハンマーでたたいて有効となるのは、顔を出した状態から、完全に出すときまでのほんの少しの時間です。この間にモグラをたたきますと、モグラはキュンと鳴いて赤くなります。モグラが100回顔を出すと、ゲームオーバーとなり、たたいた回数と当たった回数などの結果から、コメントが出ます。なお、あまりたたきすぎますと注意が出ますが、愛機のキーボードのためにもたたきすぎに注意しましょう。



プログラム

READ B\$
SPRITE \$

このゲームは、穴、モグラ、ハンマーすべてにスプライトを使用しています。スプライトのデータは880~1220行に入っており、このデータを変更することにより、オリジナルのもぐらたたきを作ることができます。

ON KEY GOSUB

モグラ、ハンマーともにスプライトを使用することで高速化ができたわけですが、実はもう一つ重要な命令、**ON KEY GOSUB** 命令を使っています。この命令は、今プログラムがどこの行を実行しようとも、ファンクションキーが押されると、この命令で指定した行に制御を移すことができるもので、いつ押されるかわからないキーに対して高速の処理を行うことが可能となります。

ON INTEVAL GOSUB
ON SPRITE GOSUB
ON STRIG GOSUB

一般にこのような処理方式を「割り込み」と呼んでいます。MSX-BASICでは、ファンクションキーによる割り込みのほか、一定時間による割り込み (**ON INTEVAL GOSUB**) スプライトの衝突による割り込み (**ON SPRITE GOSUB**) やジョイスティックによる割り込み (**ON STRIG GOSUB**) などゲームを作るのに便利な命令がたくさん用意されています。

このプログラムでは、ファンクションキーが押されると、そのキーに対応する各処理に分岐し、そのときのモグラの状態を見てモグラをうまくたたいたかどうかの判断をしています。

逆にこの方法で困る点は、ファンクションキーが押されっぱなしですと、この割り込み処理から抜け出ることができなくなり、動作がストップしてしまうことです。この点については、BASIC の特性ですのでどうしようもありません。

なお、このプログラムは16Kバイト以上のシステムならどんなものでも動作します。

```

10 ' もぐらたたき ゲーム
20 ' by Nami Aoki
30 ' debug and advice by Haruka Takagi
40 ' copyright 1984 (C) NATS
50 ' copyright 1984 (C)
60 ' Seibundo-Shinkousha publishing
70 SCREEN 0:CLS
80 PRINT "*****";
90 PRINT "*"          モグ`ラ たたき ゲーム          "*";
100 PRINT "*"          by N. Aoki & H. Takagi          "*";
110 PRINT "*"
120 PRINT "*"          copyright 1984 NATS          "*";
130 PRINT "*"          copyright 1984 (C)          "*";
140 PRINT "*"          Seibundo-Shinkousha publishing  "*";
150 PRINT "*****";
160 PRINT :PRINT
170 INPUT "          ナンイト`ハ ? (1(HARD)-3(EASY)) ";MU:SC=0:MS=
    0:FF=0
180 SCREEN 2,2,0:COLOR 15,1,7
190 ON KEY GOSUB 630,680,730,780,830
200 KEY(1) ON:KEY(2) ON:KEY(3) ON
210 KEY(4) ON:KEY(5) ON
220 OPEN "GRP:" FOR OUTPUT AS #1
230 FOR I=1 TO 5:A$=""
240 FOR J=0 TO 31
250 READ B$:A$=A$+CHR$(VAL("&H"+B$))
260 NEXT J
270 SPRITE$(I)=A$
280 NEXT I
290 X(1)=10:Y(1)=34:X(2)=50:Y(2)=54
300 X(3)=90:Y(3)=114:X(4)=140:Y(4)=64
310 X(5)=200:Y(5)=14

```

```

320 '
330 ' SCREEN SET
340 '
350 CLS:PRESET (0,0):PRINT #1,"          *** もぐら たたき ***
"
360 PUT SPRITE 20,(10,50),3,5
370 PRESET STEP(3,16):PRINT #1,"F1"
380 PUT SPRITE 21,(50,70),3,5
390 PRESET STEP(3,16):PRINT #1,"F2"
400 PUT SPRITE 22,(90,130),3,5
410 PRESET STEP(3,16):PRINT #1,"F3"
420 PUT SPRITE 23,(140,80),3,5
430 PRESET STEP(3,16):PRINT #1,"F4"
440 PUT SPRITE 24,(200,30),3,5
450 PRESET STEP(3,16):PRINT #1,"F5"
460 '
470 ' MAIN ROUTINE
480 '
490 FOR TI=0 TO MU*3:R=INT(RND(1)*5+1):NEXT
500 IF MU=3 THEN 610
510 IF MA(R)=1 THEN 530
520 MA(R)=1:MI(R)=1
530 FOR I=1 TO 5
540 IF MI(I)=0 THEN 590
550 IF MI(I)=1 THEN MI(I)=2 :PUT SPRITE 10+I,(X(I),Y(I)
),13,2:GOTO 590
560 IF MI(I)=2 THEN MI(I)=3 :PUT SPRITE 10+I,(X(I),Y(I)
),13,1:BEEP:MS=MS+1:IF MS > 99 THEN 1470 ELSE 590
570 IF MI(I)=3 THEN MI(I)=4 :PUT SPRITE 10+I,(X(I),Y(I)
),13,2:GOTO 590
580 MI(I)=0:MA(I)=0:PUT SPRITE 10+I,(X(I),209)
590 NEXT I
600 GOTO 490
610 FOR I=0 TO 5:IF MA(I)<>0 THEN 480
620 NEXT:MI(R)=1:GOTO 530

```

—リスト10(その3) もぐらたたき—

```
630 '
640 ' F1 HIT
650 FU=1:GOSUB 1260
660 IF MI(1)=3 THEN PUT SPRITE 11,(X(1),Y(1)),9,1:GOSUB
    1390
670 GOTO 1460
680 '
690 ' F2 HIT
700 FU=2:GOSUB 1260
710 IF MI(2)=3 THEN PUT SPRITE 12,(X(2),Y(2)),9,1:GOSUB
    1390
720 GOTO 1460
730 '
740 ' F3 HIT
750 FU=3:GOSUB 1260
760 IF MI(3)=3 THEN PUT SPRITE 13,(X(3),Y(3)),9,1:GOSUB
    1390
770 GOTO 1460
780 '
790 ' F4 HIT
800 FU=4:GOSUB 1260
810 IF MI(4)=3 THEN PUT SPRITE 14,(X(4),Y(4)),9,1:GOSUB
    1390
820 GOTO 1460
830 '
840 ' F5 HIT
850 FU=5:GOSUB 1260
860 IF MI(5)=3 THEN PUT SPRITE 15,(X(5),Y(5)),9,1:GOSUB
    1390
870 GOTO 1460
```

```


880 '
890 ' MOGURA
900 '
910 DATA 07,08,10,20,44,4A,CE,44
920 DATA 61,5B,40,79,42,59,60,FF
930 DATA E0,10,08,04,42,E2,A3,46
940 DATA 0A,B2,02,3E,82,32,0A,FF
950 '
960 ' MOGURA HALF
970 '
980 DATA 00,00,00,00,00,00,00,02
990 DATA 01,01,00,07,18,30,60,86
1000 DATA 00,00,00,00,00,00,00,00
1010 DATA 20,40,80,E0,18,0C,06,61
1020 '
1030 ' HAMMAR 1
1040 '
1050 DATA 1F,0F,0F,0F,0F,47,BF,BF
1060 DATA BF,BF,47,0F,0F,0F,0F,1F
1070 DATA F0,E0,E0,E0,E0,C0,FF,FF
1080 DATA FF,FF,C0,E0,E0,E0,E0,F0
1090 '
1100 ' HAMMAR 2
1110 '
1120 DATA 00,00,00,00,00,00,FF,37
1130 DATA FF,FF,00,00,00,00,00,00
1140 DATA 00,00,00,00,00,7E,BF,BF
1150 DATA BF,BF,7E,00,00,00,00,00
1160 '
1170 ' ANA
1180 '
1190 DATA 00,7F,C0,3F,7F,80,FF,00
1200 DATA 80,7F,00,80,7F,00,00,00
1210 DATA 00,FE,03,FC,FE,01,FF,00
1220 DATA 01,FE,00,01,FE,00,00,00

```

```

1230 '
1240 '
1250 '
1260 '
1270 ' HIT HAMMAR PRINT
1280 '
1290 FOR K=1TO5:KEY(K) OFF:NEXT
1300 PUT SPRITE 0, (X(FU),Y(FU)-16),11,3
1310 PUT SPRITE 1, (X(FU)+16,Y(FU)-16),11,4
1320 PLAY "L6407FG"
1330 IF PLAY(0) THEN 1330
1340 PUT SPRITE 0, (0,209)
1350 PUT SPRITE 1, (0,209)
1360 FF=FF+1
1370 PLAY "L6407DE"
1380 RETURN
1390 '
1400 ' HIT SOUND
1410 '
1420 FOR SO=0 TO 255
1430 SOUND 0,SO
1440 NEXT :SC=SC+1:RETURN
1450 MI(FU)=0:MA(FU)=0
1460 FOR K=1 TO 5 :KEY(K) ON:NEXT :RETURN


```



```

1470 ' GAME OVER
1480 '
1490 FOR K=1 TO 5:KEY(K) OFF:NEXT:SCREEN 1:LOCATE 10,5:
PRINT"♥GAME OVER♥"
1500 LOCATE 8,5:PRINT" モクラ : 100 ヒ・キ"
1510 LOCATE 8,7:PRINT" ヒット :";SC;"カイ"
1520 LOCATE 8,9:PRINT" ハンマー :";FF;"カイ"
1530 QA=SC/FF:LOCATE 3,12
1540 IF QA<=.5 THEN PRINT "まあ まあ の せいせき で"すね      ":G
OTO 1570
1550 IF QA<=.7 THEN PRINT "なか なか すは"らしい で"すね      ":G
OTO 1570
1560 PRINT "ちょうし"んてきな はんしゃしんけいて"す "
1570 IF FF>100 THEN LOCATE 3,15 :PRINT "あまり たたきすぎ"ないように
してくた"さい!!!"
1580 LOCATE 8,19:PRINT" TRY AGAIN ?";
1590 A$=INKEY$:IF A$="Y" OR A$="y" THEN RUN
1600 IF A$="N" THEN SCREEN 0:END
1610 IF PLAY(2) THEN 1590
1620 PLAY"T180O6L16ER32ER32GR32ER16 BR32ER32GR32ER","T2
00V504L4CR03GR","T200V504L4GR04DR"
1630 GOTO 1590

```



検印省略

NDC 548

MSXソフト集 Part 1

昭和60年5月10日 発行

定価 850円

編 者 N A T S

発 行 者 小 川 茂 男

発 行 所 慧誠文堂新光社

東京都千代田区神田錦町1-5-5

郵便番号 101

電話 東京 (292) 1 2 1 1 (編集)

(292) 1 2 2 1 (営業)

振替口座 東京 7-6294

印刷・広研印刷株式会社

製本・岡嶋製本工業

© 1985 NATS

Printed in Japan

(本誌掲載記事の無断転用を禁じます)

万一落丁・乱丁の場合はお取替えいたします

ISBN4-416-18449-2 C2055

本社発行
の雑誌

子供の科学／天文ガイド／MJ無線と実験／初歩のラジオ／農耕と園芸
商店界／アイデア／ブレーン／ザ・コピーライターズ／月刊 芽／ガー
デンライフ／愛犬の友／フローリスト／囲碁／DEVIC file／PORTE
FOLIO

本書のプログラムで、お気づきの点がありましたら、
手紙にて当社初ラ編集部 MSX 係宛に御一報下さい。